

Game-based cryptography in HOL

Andreas Lochbihler and S. Reza Sefidgar and Bhargav Bhatt

May 26, 2024

Abstract

In this AFP entry, we show how to specify game-based cryptographic security notions and formally prove secure several cryptographic constructions from the literature using the CryptHOL framework. Among others, we formalise the notions of a random oracle, a pseudo-random function, an unpredictable function, and of encryption schemes that are indistinguishable under chosen plaintext and/or ciphertext attacks. We prove the random-permutation/random-function switching lemma, security of the Elgamal and hashed Elgamal public-key encryption scheme and correctness and security of several constructions with pseudo-random functions.

Our proofs follow the game-hopping style advocated by Shoup [19] and Bellare and Rogaway [4], from which most of the examples have been taken. We generalise some of their results such that they can be reused in other proofs. Thanks to CryptHOL's integration with Isabelle's parametricity infrastructure, many simple hops are easily justified using the theory of representation independence.

Contents

1	Specifying security using games	3
1.1	The DDH game	3
1.2	The LCDH game	4
1.3	The IND-CCA2 game for public-key encryption	4
1.3.1	Single-user setting	6
1.3.2	Multi-user setting	7
1.4	The IND-CCA2 security for symmetric encryption schemes	8
1.5	The IND-CPA game for symmetric encryption schemes	10
1.6	The IND-CPA game for public-key encryption with oracle access	11
1.7	The IND-CPA game (public key, single instance)	13
1.8	Strongly existentially unforgeable signature scheme	14
1.8.1	Single-user setting	15
1.8.2	Multi-user setting	16
1.9	Pseudo-random function	17
1.10	Pseudo-random function	18

1.11	Random permutation	19
1.12	Reducing games with many adversary guesses to games with single guesses	20
1.13	Unpredictable function	22
2	Cryptographic constructions and their security	23
2.1	Elgamal encryption scheme	24
2.2	Hashed Elgamal in the Random Oracle Model	26
2.3	The random-permutation random-function switching lemma	29
2.4	Extending the input length of a PRF using a universal hash function	30
2.5	IND-CPA from PRF	32
2.6	IND-CCA from a PRF and an unpredictable function	34
A	Tutorial Introduction to CryptHOL	43
3	Introduction	43
3.1	Getting started	44
3.2	Getting started	45
4	Modelling cryptography using CryptHOL	45
4.1	Security notions without oracles: the CDH assumption	45
4.2	A Random Oracle	48
4.3	Cryptographic concepts: public-key encryption	49
4.4	Security notions with oracles: IND-CPA security	50
4.5	Concrete cryptographic constructions: the hashed ElGamal encryption scheme	52
5	Cryptographic proofs in CryptHOL	55
5.1	The reduction	55
5.2	Concrete security statement	57
5.3	Recording adversary queries	58
5.4	Equational program transformations	60
5.5	Capturing a failure event	62
5.6	Game hop based on a failure event	63
5.7	Optimistic sampling: the one-time-pad	66
5.8	Combining several game hops	67
6	Asymptotic security	68
6.1	Introducing a security parameter	68
6.2	Asymptotic security statements	69

1 Specifying security using games

```
theory Diffie-Hellman imports  
  CryptHOL.Cyclic-Group-SPMF  
  CryptHOL.Computational-Model  
begin
```

1.1 The DDH game

```
locale ddh =
```

```
  fixes  $\mathcal{G} :: 'grp$  cyclic-group (structure)  
begin
```

```
type-synonym 'grp' adversary = 'grp'  $\Rightarrow$  'grp'  $\Rightarrow$  'grp'  $\Rightarrow$  bool spmf
```

```
definition ddh-0 :: 'grp adversary  $\Rightarrow$  bool spmf
```

```
where ddh-0  $\mathcal{A} =$  do {  
   $x \leftarrow$  sample-uniform (order  $\mathcal{G}$ );  
   $y \leftarrow$  sample-uniform (order  $\mathcal{G}$ );  
   $\mathcal{A}$  ( $\mathbf{g} [\wedge] x$ ) ( $\mathbf{g} [\wedge] y$ ) ( $\mathbf{g} [\wedge] (x * y)$ )  
}
```

```
definition ddh-1 :: 'grp adversary  $\Rightarrow$  bool spmf
```

```
where ddh-1  $\mathcal{A} =$  do {  
   $x \leftarrow$  sample-uniform (order  $\mathcal{G}$ );  
   $y \leftarrow$  sample-uniform (order  $\mathcal{G}$ );  
   $z \leftarrow$  sample-uniform (order  $\mathcal{G}$ );  
   $\mathcal{A}$  ( $\mathbf{g} [\wedge] x$ ) ( $\mathbf{g} [\wedge] y$ ) ( $\mathbf{g} [\wedge] z$ )  
}
```

```
definition advantage :: 'grp adversary  $\Rightarrow$  real
```

```
where advantage  $\mathcal{A} = |$ spmf (ddh-0  $\mathcal{A}$ ) True  $-$  spmf (ddh-1  $\mathcal{A}$ ) True|
```

```
definition lossless :: 'grp adversary  $\Rightarrow$  bool
```

```
where lossless  $\mathcal{A} \longleftrightarrow (\forall \alpha \beta \gamma. \text{lossless-spmf } (\mathcal{A} \alpha \beta \gamma))$ 
```

```
lemma lossless-ddh-0:
```

```
   $\llbracket \text{lossless } \mathcal{A}; 0 < \text{order } \mathcal{G} \rrbracket$   
   $\implies \text{lossless-spmf } (\text{ddh-0 } \mathcal{A})$   
<proof>
```

```
lemma lossless-ddh-1:
```

```
   $\llbracket \text{lossless } \mathcal{A}; 0 < \text{order } \mathcal{G} \rrbracket$   
   $\implies \text{lossless-spmf } (\text{ddh-1 } \mathcal{A})$   
<proof>
```

```
end
```

1.2 The LCDH game

locale *lcdh* =

fixes $\mathcal{G} :: \text{'grp cyclic-group (structure)}$

begin

type-synonym *'grp' adversary* = *'grp'* \Rightarrow *'grp'* \Rightarrow *'grp'* *set spmf*

definition *lcdh* :: *'grp adversary* \Rightarrow *bool spmf*

where *lcdh* $\mathcal{A} = \text{do}$ {

$x \leftarrow \text{sample-uniform (order } \mathcal{G}\text{)}$;

$y \leftarrow \text{sample-uniform (order } \mathcal{G}\text{)}$;

$zs \leftarrow \mathcal{A} (\mathbf{g} [\wedge] x) (\mathbf{g} [\wedge] y)$;

$\text{return-spmf } (\mathbf{g} [\wedge] (x * y) \in zs)$

}

definition *advantage* :: *'grp adversary* \Rightarrow *real*

where *advantage* $\mathcal{A} = \text{spmf (lcdh } \mathcal{A}\text{) True}$

definition *lossless* :: *'grp adversary* \Rightarrow *bool*

where *lossless* $\mathcal{A} \longleftrightarrow (\forall \alpha \beta. \text{lossless-spmf } (\mathcal{A} \alpha \beta))$

lemma *lossless-lcdh*:

$\llbracket \text{lossless } \mathcal{A}; 0 < \text{order } \mathcal{G} \rrbracket$

$\implies \text{lossless-spmf (lcdh } \mathcal{A}\text{)}$

<proof>

end

end

theory *IND-CCA2 imports*

CryptHOL.Computational-Model

CryptHOL.Negligible

CryptHOL.Environment-Functor

begin

locale *pk-enc* =

fixes *key-gen* :: *security* \Rightarrow (*'ekey* \times *'dkey*) *spmf* — probabilistic

and *encrypt* :: *security* \Rightarrow *'ekey* \Rightarrow *'plain* \Rightarrow *'cipher spmf* — probabilistic

and *decrypt* :: *security* \Rightarrow *'dkey* \Rightarrow *'cipher* \Rightarrow *'plain option* — deterministic, but not used

and *valid-plain* :: *security* \Rightarrow *'plain* \Rightarrow *bool* — checks whether a plain text is valid, i.e., has the right format

1.3 The IND-CCA2 game for public-key encryption

We model an IND-CCA2 security game in the multi-user setting as described in [3].

```

locale ind-cca2 = pk-enc +
  constrains key-gen :: security ⇒ ('ekey × 'dkey) spmf
  and encrypt :: security ⇒ 'ekey ⇒ 'plain ⇒ 'cipher spmf
  and decrypt :: security ⇒ 'dkey ⇒ 'cipher ⇒ 'plain option
  and valid-plain :: security ⇒ 'plain ⇒ bool
begin

type-synonym ('ekey', 'dkey', 'cipher') state-oracle = ('ekey' × 'dkey' × 'cipher' list)
option

fun decrypt-oracle
  :: security ⇒ ('ekey, 'dkey, 'cipher) state-oracle ⇒ 'cipher
  ⇒ ('plain option × ('ekey, 'dkey, 'cipher) state-oracle) spmf
where
  decrypt-oracle η None cipher = return-spmf (None, None)
| decrypt-oracle η (Some (ekey, dkey, cstars)) cipher = return-spmf
  (if cipher ∈ set cstars then None else decrypt η dkey cipher, Some (ekey, dkey, cstars))

fun ekey-oracle
  :: security ⇒ ('ekey, 'dkey, 'cipher) state-oracle ⇒ unit ⇒ ('ekey × ('ekey, 'dkey, 'cipher)
state-oracle) spmf
where
  ekey-oracle η None - = do {
    (ekey, dkey) ← key-gen η;
    return-spmf (ekey, Some (ekey, dkey, []))
  }
| ekey-oracle η (Some (ekey, rest)) - = return-spmf (ekey, Some (ekey, rest))

lemma ekey-oracle-conv:
  ekey-oracle η σ x =
  (case σ of None ⇒ map-spmf (λ (ekey, dkey). (ekey, Some (ekey, dkey, []))) (key-gen η)
  | Some (ekey, rest) ⇒ return-spmf (ekey, Some (ekey, rest)))
  ⟨proof⟩

context notes bind-spmf-cong[fundef-cong] begin
function encrypt-oracle
  :: bool ⇒ security ⇒ ('ekey, 'dkey, 'cipher) state-oracle ⇒ 'plain × 'plain
  ⇒ ('cipher × ('ekey, 'dkey, 'cipher) state-oracle) spmf
where
  encrypt-oracle b η None m0l = do { (-, σ) ← ekey-oracle η None (); encrypt-oracle b
η σ m0l }
| encrypt-oracle b η (Some (ekey, dkey, cstars)) (m0, m1) =
  (if valid-plain η m0 ∧ valid-plain η m1 then do {
    let pb = (if b then m0 else m1);
    cstar ← encrypt η ekey pb;
    return-spmf (cstar, Some (ekey, dkey, cstar # cstars))
  } else return-spmf None)
  ⟨proof⟩
termination ⟨proof⟩

```

end

1.3.1 Single-user setting

type-synonym ('plain', 'cipher') call₁ = unit + 'cipher' + 'plain' × 'plain'

type-synonym ('ekey', 'plain', 'cipher') ret₁ = 'ekey' + 'plain' option + 'cipher'

definition oracle₁ :: bool ⇒ security

⇒ (('ekey', 'dkey', 'cipher') state-oracle, ('plain', 'cipher') call₁, ('ekey', 'plain', 'cipher') ret₁) oracle'

where oracle₁ b η = ekey-oracle η ⊕_O (decrypt-oracle η ⊕_O encrypt-oracle b η)

lemma oracle₁-simps [simp]:

oracle₁ b η s (Inl x) = map-spmf (apfst Inl) (ekey-oracle η s x)

oracle₁ b η s (Inr (Inl y)) = map-spmf (apfst (Inr ∘ Inl)) (decrypt-oracle η s y)

oracle₁ b η s (Inr (Inr z)) = map-spmf (apfst (Inr ∘ Inr)) (encrypt-oracle b η s z)

⟨proof⟩

type-synonym ('ekey', 'plain', 'cipher') adversary₁' =

(bool, ('plain', 'cipher') call₁, ('ekey', 'plain', 'cipher') ret₁) gpv

type-synonym ('ekey', 'plain', 'cipher') adversary₁ =

security ⇒ ('ekey', 'plain', 'cipher') adversary₁'

definition ind-cca2₁ :: ('ekey', 'plain', 'cipher') adversary₁ ⇒ security ⇒ bool spmf

where

ind-cca2₁ ℳ η = TRY do {

b ← coin-spmf;

(guess, s) ← exec-gpv (oracle₁ b η) (ℳ η) None;

return-spmf (guess = b)

} ELSE coin-spmf

definition advantage₁ :: ('ekey', 'plain', 'cipher') adversary₁ ⇒ advantage

where advantage₁ ℳ η = |spmf (ind-cca2₁ ℳ η) True - 1/2|

lemma advantage₁-nonneg: advantage₁ ℳ η ≥ 0 ⟨proof⟩

abbreviation secure-for₁ :: ('ekey', 'plain', 'cipher') adversary₁ ⇒ bool

where secure-for₁ ℳ ≡ negligible (advantage₁ ℳ)

definition ibounded-by₁' :: ('ekey', 'plain', 'cipher') adversary₁' ⇒ nat ⇒ bool

where ibounded-by₁' ℳ q = interaction-any-bounded-by ℳ q

abbreviation ibounded-by₁ :: ('ekey', 'plain', 'cipher') adversary₁ ⇒ (security ⇒ nat) ⇒ bool

where ibounded-by₁ ≡ rel-envir ibounded-by₁'

definition lossless₁' :: ('ekey', 'plain', 'cipher') adversary₁' ⇒ bool

where lossless₁' ℳ = lossless-gpv ℳ-full ℳ

abbreviation $lossless_1 :: ('ekey, 'plain, 'cipher) adversary_1 \Rightarrow bool$
where $lossless_1 \equiv pred\text{-}envir\ lossless_1'$

lemma $lossless\text{-}decrypt\text{-}oracle$ [simp]: $lossless\text{-}spmf (decrypt\text{-}oracle\ \eta\ \sigma\ ciph)$
 $\langle proof \rangle$

lemma $lossless\text{-}ekey\text{-}oracle$ [simp]:
 $lossless\text{-}spmf (ekey\text{-}oracle\ \eta\ \sigma\ x) \longleftrightarrow (\sigma = None \longrightarrow lossless\text{-}spmf (key\text{-}gen\ \eta))$
 $\langle proof \rangle$

lemma $lossless\text{-}encrypt\text{-}oracle$ [simp]:
 $\llbracket \sigma = None \implies lossless\text{-}spmf (key\text{-}gen\ \eta);$
 $\wedge ekey\ m.\ valid\text{-}plain\ \eta\ m \implies lossless\text{-}spmf (encrypt\ \eta\ ekey\ m) \rrbracket$
 $\implies lossless\text{-}spmf (encrypt\text{-}oracle\ b\ \eta\ \sigma\ (m0, m1)) \longleftrightarrow valid\text{-}plain\ \eta\ m0 \wedge valid\text{-}plain$
 $\eta\ m1$
 $\langle proof \rangle$

1.3.2 Multi-user setting

definition $oracle_n :: bool \Rightarrow security$
 $\Rightarrow ('i \Rightarrow ('ekey, 'dkey, 'cipher) state\text{-}oracle, 'i \times ('plain, 'cipher) call_1, ('ekey, 'plain,$
 $'cipher) ret_1) oracle'$
where $oracle_n\ b\ \eta = family\text{-}oracle (\lambda\cdot. oracle_1\ b\ \eta)$

lemma $oracle_n\text{-}apply$ [simp]:
 $oracle_n\ b\ \eta\ s\ (i, x) = map\text{-}spmf (apsnd (fun\text{-}upd\ s\ i)) (oracle_1\ b\ \eta\ (s\ i)\ x)$
 $\langle proof \rangle$

type-synonym $('i, 'ekey, 'plain, 'cipher) adversary_n' =$
 $(bool, 'i \times ('plain, 'cipher) call_1, ('ekey, 'plain, 'cipher) ret_1) gpv$

type-synonym $('i, 'ekey, 'plain, 'cipher) adversary_n =$
 $security \Rightarrow ('i, 'ekey, 'plain, 'cipher) adversary_n'$

definition $ind\text{-}cca2_n :: ('i, 'ekey, 'plain, 'cipher) adversary_n \Rightarrow security \Rightarrow bool\ spmf$
where

$ind\text{-}cca2_n\ \mathcal{A}\ \eta = TRY\ do\ \{$
 $b \leftarrow coin\text{-}spmf;$
 $(guess, \sigma) \leftarrow exec\text{-}gpv (oracle_n\ b\ \eta) (\mathcal{A}\ \eta) (\lambda\cdot. None);$
 $return\text{-}spmf (guess = b)$
 $\} ELSE\ coin\text{-}spmf$

definition $advantage_n :: ('i, 'ekey, 'plain, 'cipher) adversary_n \Rightarrow advantage$
where $advantage_n\ \mathcal{A}\ \eta = |spmf (ind\text{-}cca2_n\ \mathcal{A}\ \eta)\ True - 1/2|$

lemma $advantage_n\text{-}nonneg$: $advantage_n\ \mathcal{A}\ \eta \geq 0$ $\langle proof \rangle$

abbreviation $secure\text{-}for_n :: ('i, 'ekey, 'plain, 'cipher) adversary_n \Rightarrow bool$
where $secure\text{-}for_n\ \mathcal{A} \equiv negligible (advantage_n\ \mathcal{A})$

definition $\text{ibounded-by}_n' :: ('i, 'ekey, 'plain, 'cipher) \text{adversary}_n' \Rightarrow \text{nat} \Rightarrow \text{bool}$
where $\text{ibounded-by}_n' \mathcal{A} q = \text{interaction-any-bounded-by } \mathcal{A} q$

abbreviation $\text{ibounded-by}_n :: ('i, 'ekey, 'plain, 'cipher) \text{adversary}_n \Rightarrow (\text{security} \Rightarrow \text{nat}) \Rightarrow \text{bool}$
where $\text{ibounded-by}_n \equiv \text{rel-envir } \text{ibounded-by}_n'$

definition $\text{lossless}_n' :: ('i, 'ekey, 'plain, 'cipher) \text{adversary}_n' \Rightarrow \text{bool}$
where $\text{lossless}_n' \mathcal{A} = \text{lossless-gpv } \mathcal{I}\text{-full } \mathcal{A}$

abbreviation $\text{lossless}_n :: ('i, 'ekey, 'plain, 'cipher) \text{adversary}_n \Rightarrow \text{bool}$
where $\text{lossless}_n \equiv \text{pred-envir } \text{lossless}_n'$

definition $\text{cipher-queries} :: ('i \Rightarrow ('ekey, 'dkey, 'cipher) \text{state-oracle}) \Rightarrow 'cipher \text{ set}$
where $\text{cipher-queries } ose = (\bigcup(-, -, \text{ciphers}) \in \text{ran } ose. \text{set } \text{ciphers})$

lemma cipher-queriesI :
 $\llbracket ose\ n = \text{Some } (ek, dk, \text{ciphers}); x \in \text{set } \text{ciphers} \rrbracket \Longrightarrow x \in \text{cipher-queries } ose$
 $\langle \text{proof} \rangle$

lemma cipher-queriesE :
assumes $x \in \text{cipher-queries } ose$
obtains $(\text{cipher-queries})\ n\ ek\ dk\ \text{ciphers}$ **where** $ose\ n = \text{Some } (ek, dk, \text{ciphers})\ x \in \text{set } \text{ciphers}$
 $\langle \text{proof} \rangle$

lemma $\text{cipher-queries-updE}$:
assumes $x \in \text{cipher-queries } (ose(n \mapsto (ek, dk, \text{ciphers})))$
obtains $(old)\ x \in \text{cipher-queries } ose\ x \notin \text{set } \text{ciphers} \mid (new)\ x \in \text{set } \text{ciphers}$
 $\langle \text{proof} \rangle$

lemma $\text{cipher-queries-empty}$ [simp]: $\text{cipher-queries } \text{Map.empty} = \{\}$
 $\langle \text{proof} \rangle$

end

end

1.4 The IND-CCA2 security for symmetric encryption schemes

theory IND-CCA2-sym **imports**
 $\text{CryptHOL.Computational-Model}$
begin

locale $\text{ind-cca} =$
fixes $\text{key-gen} :: 'key\ \text{spmf}$
and $\text{encrypt} :: 'key \Rightarrow 'message \Rightarrow 'cipher\ \text{spmf}$
and $\text{decrypt} :: 'key \Rightarrow 'cipher \Rightarrow 'message\ \text{option}$

and *msg-predicate* :: 'message \Rightarrow bool
begin

type-synonym ('message', 'cipher') *adversary* =
 (bool, 'message' \times 'message' + 'cipher', 'cipher' option + 'message' option) *gpv*

definition *oracle-encrypt* :: 'key \Rightarrow bool \Rightarrow ('message \times 'message, 'cipher option, 'cipher set) *callee*

where

oracle-encrypt *k b L* = (λ (*msg1*, *msg0*).
 (case *msg-predicate* *msg1* \wedge *msg-predicate* *msg0* of
 True \Rightarrow do {
 c \leftarrow *encrypt* *k* (if *b* then *msg1* else *msg0*);
 return-spmf (Some *c*, {*c*} \cup *L*)
 }
 | False \Rightarrow return-spmf (None, *L*)))

lemma *lossless-oracle-encrypt* [*simp*]:

assumes *lossless-spmf* (*encrypt* *k m1*) **and** *lossless-spmf* (*encrypt* *k m0*)

shows *lossless-spmf* (*oracle-encrypt* *k b L* (*m1*, *m0*))

<proof>

definition *oracle-decrypt* :: 'key \Rightarrow ('cipher, 'message option, 'cipher set) *callee*

where *oracle-decrypt* *k L c* = return-spmf (if *c* \in *L* then None else *decrypt* *k c*, *L*)

lemma *lossless-oracle-decrypt* [*simp*]: *lossless-spmf* (*oracle-decrypt* *k L c*)

<proof>

definition *game* :: ('message, 'cipher) *adversary* \Rightarrow bool *spmf*

where

game *A* = do {
 key \leftarrow *key-gen*;
 b \leftarrow *coin-spmf*;
 (*b'*, *L'*) \leftarrow *exec-gpv* (*oracle-encrypt* *key b* \oplus_O *oracle-decrypt* *key*) *A* {};
 return-spmf (*b = b'*)
 }

definition *advantage* :: ('message, 'cipher) *adversary* \Rightarrow real

where *advantage* *A* = |*spmf* (*game* *A*) True - 1 / 2|

lemma *advantage-nonneg*: $0 \leq$ *advantage* *A* *<proof>*

end

end

theory *IND-CPA imports*

CryptHOL.Generative-Probabilistic-Value

CryptHOL.Computational-Model
CryptHOL.Negligible
begin

1.5 The IND-CPA game for symmetric encryption schemes

locale *ind-cpa* =
fixes *key-gen* :: 'key *spmf* — probabilistic
and *encrypt* :: 'key ⇒ 'plain ⇒ 'cipher *spmf* — probabilistic
and *decrypt* :: 'key ⇒ 'cipher ⇒ 'plain *option* — deterministic, but not used
and *valid-plain* :: 'plain ⇒ *bool* — checks whether a plain text is valid, i.e., has the right format
begin

We cannot incorporate the predicate *valid-plain* in the type 'plain of plaintexts, because the single 'plain must contain plaintexts for all values of the security parameter, as HOL does not have dependent types. Consequently, the oracle has to ensure that the received plaintexts are valid.

type-synonym ('plain', 'cipher', 'state) *adversary* =
 (('plain' × 'plain') × 'state, 'plain', 'cipher') *gpv*
 × ('cipher' ⇒ 'state ⇒ (*bool*, 'plain', 'cipher') *gpv*)

definition *encrypt-oracle* :: 'key ⇒ *unit* ⇒ 'plain ⇒ ('cipher × *unit*) *spmf*
where
encrypt-oracle *key* σ *plain* = *do* {
 cipher ← *encrypt* *key* *plain*;
 return-spmf (*cipher*, ())
}

definition *ind-cpa* :: ('plain', 'cipher', 'state) *adversary* ⇒ *bool* *spmf*
where
ind-cpa \mathcal{A} = *do* {
 let ($\mathcal{A}1$, $\mathcal{A}2$) = \mathcal{A} ;
 key ← *key-gen*;
 b ← *coin-spmf*;
 (*guess*, -) ← *exec-gpv* (*encrypt-oracle* *key*) (*do* {
 (*m0*, *m1*), σ ← $\mathcal{A}1$;
 if *valid-plain* *m0* ∧ *valid-plain* *m1* *then* *do* {
 cipher ← *lift-spmf* (*encrypt* *key* (*if* *b* *then* *m0* *else* *m1*));
 $\mathcal{A}2$ *cipher* σ
 } *else* *lift-spmf* *coin-spmf*
 }) ();
 return-spmf (*guess* = *b*)
}

definition *advantage* :: ('plain', 'cipher', 'state) *adversary* ⇒ *real*
where *advantage* \mathcal{A} = |*spmf* (*ind-cpa* \mathcal{A}) *True* − 1/2|

lemma *advantage-nonneg*: *advantage* \mathcal{A} ≥ 0 ⟨*proof*⟩

definition *ibounded-by* :: ('plain, 'cipher, 'state) adversary \Rightarrow enat \Rightarrow bool

where

ibounded-by = $(\lambda(\mathcal{A}1, \mathcal{A}2) q.$
 $(\exists q1\ q2. \text{interaction-any-bounded-by } \mathcal{A}1\ q1 \wedge (\forall \text{cipher } \sigma. \text{interaction-any-bounded-by}$
 $(\mathcal{A}2\ \text{cipher } \sigma)\ q2) \wedge q1 + q2 \leq q))$

lemma *ibounded-byE* [consumes 1, case-names *ibounded-by*, *elim?*]:

assumes *ibounded-by* ($\mathcal{A}1, \mathcal{A}2$) q

obtains $q1\ q2$

where $q1 + q2 \leq q$

and *interaction-any-bounded-by* $\mathcal{A}1\ q1$

and $\bigwedge \text{cipher } \sigma. \text{interaction-any-bounded-by } (\mathcal{A}2\ \text{cipher } \sigma)\ q2$

<proof>

lemma *ibounded-byI* [intro?]:

$\llbracket \text{interaction-any-bounded-by } \mathcal{A}1\ q1; \bigwedge \text{cipher } \sigma. \text{interaction-any-bounded-by } (\mathcal{A}2\ \text{cipher } \sigma)\ q2; q1 + q2 \leq q \rrbracket$

$\implies \text{ibounded-by } (\mathcal{A}1, \mathcal{A}2)\ q$

<proof>

definition *lossless* :: ('plain, 'cipher, 'state) adversary \Rightarrow bool

where *lossless* = $(\lambda(\mathcal{A}1, \mathcal{A}2). \text{lossless-gpv } \mathcal{I}\text{-full } \mathcal{A}1 \wedge (\forall \text{cipher } \sigma. \text{lossless-gpv } \mathcal{I}\text{-full}$
 $(\mathcal{A}2\ \text{cipher } \sigma)))$

end

end

theory *IND-CPA-PK imports*

CryptHOL.Computational-Model

CryptHOL.Negligible

begin

1.6 The IND-CPA game for public-key encryption with oracle access

locale *ind-cpa-pk* =

fixes *key-gen* :: ('pubkey \times 'privkey, 'call, 'ret) gpv — probabilistic

and *aencrypt* :: 'pubkey \Rightarrow 'plain \Rightarrow ('cipher, 'call, 'ret) gpv — probabilistic w/ access to an oracle

and *adecrypt* :: 'privkey \Rightarrow 'cipher \Rightarrow ('plain, 'call, 'ret) gpv — not used

and *valid-plains* :: 'plain \Rightarrow 'plain \Rightarrow bool — checks whether a pair of plaintexts is valid, i.e., they have the right format

begin

We cannot incorporate the predicate *valid-plain* in the type 'plain of plaintexts, because the single 'plain must contain plaintexts for all values of the security parameter, as HOL does not have dependent types. Consequently, the game has to

ensure that the received plaintexts are valid.

type-synonym ('pubkey', 'plain', 'cipher', 'call', 'ret', 'state) adversary =
 ('pubkey' \Rightarrow (('plain' \times 'plain') \times 'state', 'call', 'ret') gpv)
 \times ('cipher' \Rightarrow 'state' \Rightarrow (bool, 'call', 'ret') gpv)

fun ind-cpa :: ('pubkey', 'plain', 'cipher', 'call', 'ret', 'state) adversary \Rightarrow (bool, 'call', 'ret') gpv

where

ind-cpa ($\mathcal{A}1$, $\mathcal{A}2$) = TRY do {
 (pk, sk) \leftarrow key-gen;
 b \leftarrow lift-spmf coin-spmf;
 ((m0, m1), σ) \leftarrow ($\mathcal{A}1$ pk);
 assert-gpv (valid-plains m0 m1);
 cipher \leftarrow aencrypt pk (if b then m0 else m1);
 guess \leftarrow $\mathcal{A}2$ cipher σ ;
 Done (guess = b)
} ELSE lift-spmf coin-spmf

definition advantage :: (' σ \Rightarrow 'call' \Rightarrow ('ret' \times ' σ ') spmf) \Rightarrow ' σ ' \Rightarrow ('pubkey', 'plain', 'cipher', 'call', 'ret', 'state) adversary \Rightarrow real

where advantage oracle σ \mathcal{A} = |spmf (run-gpv oracle (ind-cpa \mathcal{A}) σ) True - 1/2|

lemma advantage-nonneg: advantage oracle σ $\mathcal{A} \geq 0$ <proof>

definition ibounded-by :: ('call' \Rightarrow bool) \Rightarrow ('pubkey', 'plain', 'cipher', 'call', 'ret', 'state) adversary \Rightarrow enat \Rightarrow bool

where

ibounded-by consider = (λ ($\mathcal{A}1$, $\mathcal{A}2$) q.
 (\exists q1 q2. (\forall pk. interaction-bounded-by consider ($\mathcal{A}1$ pk) q1) \wedge (\forall cipher σ . interaction-bounded-by consider ($\mathcal{A}2$ cipher σ) q2) \wedge q1 + q2 \leq q))

lemma ibounded-by'E [consumes 1, case-names ibounded-by', elim?]:

assumes ibounded-by consider ($\mathcal{A}1$, $\mathcal{A}2$) q

obtains q1 q2

where q1 + q2 \leq q

and \wedge pk. interaction-bounded-by consider ($\mathcal{A}1$ pk) q1

and \wedge cipher σ . interaction-bounded-by consider ($\mathcal{A}2$ cipher σ) q2

<proof>

lemma ibounded-byI [intro?]:

$\llbracket \wedge$ pk. interaction-bounded-by consider ($\mathcal{A}1$ pk) q1; \wedge cipher σ . interaction-bounded-by consider ($\mathcal{A}2$ cipher σ) q2; q1 + q2 \leq q \rrbracket

\implies ibounded-by consider ($\mathcal{A}1$, $\mathcal{A}2$) q

<proof>

definition lossless :: ('pubkey', 'plain', 'cipher', 'call', 'ret', 'state) adversary \Rightarrow bool

where lossless = (λ ($\mathcal{A}1$, $\mathcal{A}2$). (\forall pk. lossless-gpv \mathcal{I} -full ($\mathcal{A}1$ pk)) \wedge (\forall cipher σ . lossless-gpv \mathcal{I} -full ($\mathcal{A}2$ cipher σ)))

end

end

theory *IND-CPA-PK-Single* **imports**
 CryptHOL.Computational-Model
begin

1.7 The IND-CPA game (public key, single instance)

locale *ind-cpa* =
 fixes *key-gen* :: ('pub-key × 'priv-key) *spmf* — probabilistic
 and *aencrypt* :: 'pub-key ⇒ 'plain ⇒ 'cipher *spmf* — probabilistic
 and *adecrypt* :: 'priv-key ⇒ 'cipher ⇒ 'plain *option* — deterministic, but not used
 and *valid-plains* :: 'plain ⇒ 'plain ⇒ *bool* — checks whether a pair of plaintexts is valid,
 i.e., they both have the right format
begin

We cannot incorporate the predicate *valid-plain* in the type *'plain* of plaintexts, because the single *'plain* must contain plaintexts for all values of the security parameter, as HOL does not have dependent types. Consequently, the oracle has to ensure that the received plaintexts are valid.

type-synonym ('pub-key', 'plain', 'cipher', 'state) *adversary* =
 ('pub-key' ⇒ (('plain' × 'plain') × 'state) *spmf*)
 × ('cipher' ⇒ 'state ⇒ *bool* *spmf*)

primrec *ind-cpa* :: ('pub-key, 'plain, 'cipher, 'state) *adversary* ⇒ *bool* *spmf*

where

ind-cpa (*ℳ1*, *ℳ2*) = *TRY* *do* {
 (*pk*, *sk*) ← *key-gen*;
 ((*m0*, *m1*), *σ*) ← *ℳ1* *pk*;
 - :: *unit* ← *assert-spmf* (*valid-plains* *m0* *m1*);
 b ← *coin-spmf*;
 cipher ← *aencrypt* *pk* (if *b* then *m0* else *m1*);
 b' ← *ℳ2* *cipher* *σ*;
 return-spmf (*b* = *b'*)
} *ELSE* *coin-spmf*

declare *ind-cpa.simps* [*simp del*]

definition *advantage* :: ('pub-key, 'plain, 'cipher, 'state) *adversary* ⇒ *real*
where *advantage* *ℳ* = |*spmf* (*ind-cpa* *ℳ*) *True* - 1/2|

definition *lossless* :: ('pub-key, 'plain, 'cipher, 'state) *adversary* ⇒ *bool*
where

lossless *ℳ* \longleftrightarrow
 ((\forall *pk*. *lossless-spmf* (*fst* *ℳ* *pk*)) \wedge
 (\forall *cipher* *σ*. *lossless-spmf* (*snd* *ℳ* *cipher* *σ*)))

lemma *lossless-ind-cpa*:
 $\llbracket \text{lossless } \mathcal{A}; \text{lossless-spmf } (\text{key-gen}) \rrbracket \implies \text{lossless-spmf } (\text{ind-cpa } \mathcal{A})$
 <proof>

end

end

theory *SUF-CMA imports*
CryptHOL.Computational-Model
CryptHOL.Negligible
CryptHOL.Environment-Functor
begin

1.8 Strongly existentially unforgeable signature scheme

locale *sig-scheme* =
fixes *key-gen* :: $\text{security} \Rightarrow ('vkey \times 'sigkey) \text{ spmf}$
and *sign* :: $\text{security} \Rightarrow 'sigkey \Rightarrow 'message \Rightarrow 'signature \text{ spmf}$
and *verify* :: $\text{security} \Rightarrow 'vkey \Rightarrow 'message \Rightarrow 'signature \Rightarrow \text{bool}$ — verification is deterministic
and *valid-message* :: $\text{security} \Rightarrow 'message \Rightarrow \text{bool}$

locale *suf-cma* = *sig-scheme* +
constrains *key-gen* :: $\text{security} \Rightarrow ('vkey \times 'sigkey) \text{ spmf}$
and *sign* :: $\text{security} \Rightarrow 'sigkey \Rightarrow 'message \Rightarrow 'signature \text{ spmf}$
and *verify* :: $\text{security} \Rightarrow 'vkey \Rightarrow 'message \Rightarrow 'signature \Rightarrow \text{bool}$
and *valid-message* :: $\text{security} \Rightarrow 'message \Rightarrow \text{bool}$
begin

type-synonym (*'vkey'*, *'sigkey'*, *'message'*, *'signature'*) *state-oracle*
 = (*'vkey'* \times *'sigkey'* \times (*'message'* \times *'signature'*) *list*) *option*

fun *vkey-oracle* :: $\text{security} \Rightarrow ((\text{'vkey'}, \text{'sigkey'}, \text{'message'}, \text{'signature'}) \text{ state-oracle}, \text{unit}, \text{'vkey'}) \text{ oracle}'$

where

vkey-oracle η *None* - = *do* {
 (*vkey*, *sigkey*) \leftarrow *key-gen* η ;
return-spmf (*vkey*, *Some* (*vkey*, *sigkey*, []))
 }
 | \wedge *log*. *vkey-oracle* η (*Some* (*vkey*, *sigkey*, *log*)) - = *return-spmf* (*vkey*, *Some* (*vkey*, *sigkey*, *log*))

context notes *bind-spmf-cong*[*fundef-cong*] **begin**

function *sign-oracle*

:: $\text{security} \Rightarrow ((\text{'vkey'}, \text{'sigkey'}, \text{'message'}, \text{'signature'}) \text{ state-oracle}, \text{'message'}, \text{'signature'}) \text{ oracle}'$

where

$sign\text{-oracle } \eta \text{ None } m = do \{ (-, \sigma) \leftarrow vkey\text{-oracle } \eta \text{ None } (); sign\text{-oracle } \eta \sigma m \}$
 $| \wedge log. sign\text{-oracle } \eta (Some (vkey, skey, log)) m =$
 $(if \text{ valid-message } \eta m \text{ then do } \{$
 $sig \leftarrow sign \eta skey m;$
 $return\text{-spmf } (sig, Some (vkey, skey, (m, sig) \# log))$
 $\} \text{ else return-pmf None})$
 $\langle proof \rangle$
termination $\langle proof \rangle$
end

lemma *lossless-vkey-oracle* [simp]:

$lossless\text{-spmf } (vkey\text{-oracle } \eta \sigma x) \longleftrightarrow (\sigma = None \longrightarrow lossless\text{-spmf } (key\text{-gen } \eta))$
 $\langle proof \rangle$

lemma *lossless-sign-oracle* [simp]:

$\llbracket \sigma = None \implies lossless\text{-spmf } (key\text{-gen } \eta);$
 $\wedge skey m. \text{ valid-message } \eta m \implies lossless\text{-spmf } (sign \eta skey m) \rrbracket$
 $\implies lossless\text{-spmf } (sign\text{-oracle } \eta \sigma m) \longleftrightarrow \text{ valid-message } \eta m$
 $\langle proof \rangle$

lemma *lossless-sign-oracle-Some: fixes log shows*

$lossless\text{-spmf } (sign\text{-oracle } \eta (Some (vkey, skey, log)) m) \longleftrightarrow lossless\text{-spmf } (sign \eta skey$
 $m) \wedge \text{ valid-message } \eta m$
 $\langle proof \rangle$

1.8.1 Single-user setting

type-synonym *'message'* $call_1 = unit + 'message'$

type-synonym *('vkey', 'signature')* $ret_1 = 'vkey' + 'signature'$

definition *oracle₁* :: *security*

$\Rightarrow (('vkey, 'sigkey, 'message, 'signature) \text{ state-oracle}, 'message \text{ call}_1, ('vkey, 'signature)$
 $ret_1) \text{ oracle}'$

where $oracle_1 \eta = vkey\text{-oracle } \eta \oplus_O sign\text{-oracle } \eta$

lemma *oracle₁-simps* [simp]:

$oracle_1 \eta s (Inl x) = map\text{-spmf } (apfst Inl) (vkey\text{-oracle } \eta s x)$
 $oracle_1 \eta s (Inr y) = map\text{-spmf } (apfst Inr) (sign\text{-oracle } \eta s y)$
 $\langle proof \rangle$

type-synonym *('vkey', 'message', 'signature')* $adversary_1' =$

$(('message' \times 'signature'), 'message' \text{ call}_1, ('vkey, 'signature') ret_1) \text{ gpv}$

type-synonym *('vkey', 'message', 'signature')* $adversary_1 =$

$security \Rightarrow ('vkey, 'message, 'signature') \text{ adversary}_1'$

definition *suf-cma₁* :: *('vkey', 'message', 'signature')* $adversary_1 \Rightarrow security \Rightarrow bool \text{ spmf}$

where

$\wedge log. \text{ suf-cma}_1 \mathcal{A} \eta = do \{$

```

((m, sig),  $\sigma$ )  $\leftarrow$  exec-gpv (oracle1  $\eta$ ) ( $\mathcal{A}$   $\eta$ ) None;
return-spmf (
  case  $\sigma$  of None  $\Rightarrow$  False
  | Some (vkey, skey, log)  $\Rightarrow$  verify  $\eta$  vkey m sig  $\wedge$  (m, sig)  $\notin$  set log)
}

```

definition advantage₁ :: ('vkey, 'message, 'signature) adversary₁ \Rightarrow advantage
where advantage₁ \mathcal{A} η = spmf (suf-cma₁ \mathcal{A} η) True

lemma advantage₁-nonneg: advantage₁ \mathcal{A} η \geq 0 (proof)

abbreviation secure-for₁ :: ('vkey, 'message, 'signature) adversary₁ \Rightarrow bool
where secure-for₁ \mathcal{A} \equiv negligible (advantage₁ \mathcal{A})

definition ibounded-by₁' :: ('vkey, 'message, 'signature) adversary₁' \Rightarrow nat \Rightarrow bool
where ibounded-by₁' \mathcal{A} q = (interaction-any-bounded-by \mathcal{A} q)

abbreviation ibounded-by₁ :: ('vkey, 'message, 'signature) adversary₁ \Rightarrow (security \Rightarrow nat)
 \Rightarrow bool
where ibounded-by₁ \equiv rel-envir ibounded-by₁'

definition lossless₁' :: ('vkey, 'message, 'signature) adversary₁' \Rightarrow bool
where lossless₁' \mathcal{A} = (lossless-gpv \mathcal{I} -full \mathcal{A})

abbreviation lossless₁ :: ('vkey, 'message, 'signature) adversary₁ \Rightarrow bool
where lossless₁ \equiv pred-envir lossless₁'

1.8.2 Multi-user setting

definition oracle_n :: security
 \Rightarrow (i \Rightarrow ('vkey, 'sigkey, 'message, 'signature) state-oracle, 'i \times 'message call₁, ('vkey, 'signature) ret₁) oracle'
where oracle_n η = family-oracle (λ -. oracle₁ η)

lemma oracle_n-apply [simp]:
oracle_n η s (i, x) = map-spmf (apsnd (fun-upd s i)) (oracle₁ η (s i) x)
(proof)

type-synonym ('i, 'vkey', 'message', 'signature') adversary_n' =
(('i \times 'message' \times 'signature'), 'i \times 'message' call₁, ('vkey', 'signature') ret₁) gpv

type-synonym ('i, 'vkey', 'message', 'signature') adversary_n =
security \Rightarrow ('i, 'vkey', 'message', 'signature') adversary_n'

definition suf-cma_n :: ('i, 'vkey, 'message, 'signature) adversary_n \Rightarrow security \Rightarrow bool spmf
where

```

 $\wedge$ log. suf-cman  $\mathcal{A}$   $\eta$  = do {
  ((i, m, sig),  $\sigma$ )  $\leftarrow$  exec-gpv (oraclen  $\eta$ ) ( $\mathcal{A}$   $\eta$ ) ( $\lambda$ -. None);
  return-spmf (
    case  $\sigma$  i of None  $\Rightarrow$  False

```


} | *Some* (vkey, skey, log) \Rightarrow *verify* η vkey m sig \wedge (m, sig) \notin set log
}

definition *advantage_n* :: ('i, 'vkey, 'message, 'signature) *adversary_n* \Rightarrow *advantage*
where *advantage_n* \mathcal{A} η = *spmf* (*suf-cma_n* \mathcal{A} η) *True*

lemma *advantage_n-nonneg*: *advantage_n* \mathcal{A} η \geq 0 *(proof)*

abbreviation *secure-for_n* :: ('i, 'vkey, 'message, 'signature) *adversary_n* \Rightarrow *bool*
where *secure-for_n* \mathcal{A} \equiv *negligible* (*advantage_n* \mathcal{A})

definition *ibounded-by_n'* :: ('i, 'vkey, 'message, 'signature) *adversary_n'* \Rightarrow *nat* \Rightarrow *bool*
where *ibounded-by_n'* \mathcal{A} q = (*interaction-any-bounded-by* \mathcal{A} q)

abbreviation *ibounded-by_n* :: ('i, 'vkey, 'message, 'signature) *adversary_n* \Rightarrow (*security* \Rightarrow *nat*) \Rightarrow *bool*
where *ibounded-by_n* \equiv *rel-envir* *ibounded-by_n'*

definition *lossless_n'* :: ('i, 'vkey, 'message, 'signature) *adversary_n'* \Rightarrow *bool*
where *lossless_n'* \mathcal{A} = (*lossless-gpv* \mathcal{S} -full \mathcal{A})

abbreviation *lossless_n* :: ('i, 'vkey, 'message, 'signature) *adversary_n* \Rightarrow *bool*
where *lossless_n* \equiv *pred-envir* *lossless_n'*

end

end

theory *Pseudo-Random-Function* **imports**

CryptHOL.Computational-Model

begin

1.9 Pseudo-random function

locale *random-function* =

fixes p :: 'a *spmf*

begin

type-synonym ('b, 'a) *dict* = 'b \rightarrow 'a'

definition *random-oracle* :: ('b, 'a) *dict* \Rightarrow 'b \Rightarrow ('a \times ('b, 'a) *dict*) *spmf*

where

random-oracle σ x =

(*case* σ x of *Some* y \Rightarrow *return-spmf* (y, σ)

| *None* \Rightarrow p \gg (λ y. *return-spmf* (y, σ (x \mapsto y))))

definition *forgetful-random-oracle* :: *unit* \Rightarrow 'b \Rightarrow ('a \times *unit*) *spmf*

where

forgetful-random-oracle $\sigma x = p \gg= (\lambda y. \text{return-spmf } (y, ()))$

lemma *weight-random-oracle* [simp]:

weight-spmf $p = 1 \implies \text{weight-spmf } (\text{random-oracle } \sigma x) = 1$
<proof>

lemma *lossless-random-oracle* [simp]:

lossless-spmf $p \implies \text{lossless-spmf } (\text{random-oracle } \sigma x)$
<proof>

sublocale *finite*: *callee-invariant-on random-oracle* $\lambda \sigma. \text{finite } (\text{dom } \sigma) \mathcal{I}\text{-full}$
<proof>

lemma *card-dom-random-oracle*:

assumes *interaction-any-bounded-by* $\mathcal{A} q$
and $(y, \sigma') \in \text{set-spmf } (\text{exec-gpv } \text{random-oracle } \mathcal{A} \sigma)$
and *fin*: *finite* $(\text{dom } \sigma)$
shows $\text{card } (\text{dom } \sigma') \leq q + \text{card } (\text{dom } \sigma)$
<proof>

end

1.10 Pseudo-random function

locale *prf* =

fixes *key-gen* :: *'key* *spmf*
and *prf* :: *'key* \Rightarrow *'domain* \Rightarrow *'range*
and *rand* :: *'range* *spmf*
begin

sublocale *random-function* *rand* *<proof>*

definition *prf-oracle* :: *'key* \Rightarrow *unit* \Rightarrow *'domain* \Rightarrow (*'range* \times *unit*) *spmf*
where *prf-oracle* *key* $\sigma x = \text{return-spmf } (\text{prf } \text{key } x, ())$

type-synonym (*'domain'*, *'range'*) *adversary* = (*bool*, *'domain'*, *'range'*) *gpv*

definition *game-0* :: (*'domain'*, *'range'*) *adversary* \Rightarrow *bool* *spmf*

where

game-0 $\mathcal{A} = \text{do } \{$
 key \leftarrow *key-gen*;
 (*b*, -) \leftarrow *exec-gpv* (*prf-oracle* *key*) $\mathcal{A} ()$;
 return-spmf *b*
}

definition *game-1* :: (*'domain'*, *'range'*) *adversary* \Rightarrow *bool* *spmf*

where

game-1 $\mathcal{A} = \text{do } \{$
 (*b*, -) \leftarrow *exec-gpv* *random-oracle* $\mathcal{A} \text{Map.empty}$;

```

    return-spmf b
  }

```

definition *advantage* :: ('domain, 'range) adversary \Rightarrow real
where *advantage* $\mathcal{A} = | \text{spm}f(\text{game-0 } \mathcal{A}) \text{ True} - \text{spm}f(\text{game-1 } \mathcal{A}) \text{ True} |$

lemma *advantage-nonneg*: *advantage* $\mathcal{A} \geq 0$
 <proof>

abbreviation *lossless* :: ('domain, 'range) adversary \Rightarrow bool
where *lossless* \equiv *lossless-gpv* \mathcal{I} -full

abbreviation (*input*) *ibounded-by* :: ('domain, 'range) adversary \Rightarrow enat \Rightarrow bool
where *ibounded-by* \equiv *interaction-any-bounded-by*

end

end

1.11 Random permutation

theory *Pseudo-Random-Permutation* **imports**

CryptHOL.Computational-Model

begin

locale *random-permutation* =

fixes *A* :: 'b set

begin

definition *random-permutation* :: ('a \rightarrow 'b) \Rightarrow 'a \Rightarrow ('b \times ('a \rightarrow 'b)) spmf
where

random-permutation σ *x* =

(case σ *x* of Some *y* \Rightarrow return-spmf (*y*, σ))

| None \Rightarrow spmf-of-set (*A* - ran σ) $\gg\equiv$ ($\lambda y.$ return-spmf (*y*, $\sigma(x \mapsto y)$)))

lemma *weight-random-oracle* [simp]:

$\llbracket \text{finite } A; A - \text{ran } \sigma \neq \{\} \rrbracket \Longrightarrow \text{weight-spmf } (\text{random-permutation } \sigma \ x) = 1$
 <proof>

lemma *lossless-random-oracle* [simp]:

$\llbracket \text{finite } A; A - \text{ran } \sigma \neq \{\} \rrbracket \Longrightarrow \text{lossless-spmf } (\text{random-permutation } \sigma \ x)$
 <proof>

sublocale *finite*: *callee-invariant-on random-permutation* $\lambda \sigma.$ *finite* (dom σ) \mathcal{I} -full
 <proof>

lemma *card-dom-random-oracle*:

assumes *interaction-any-bounded-by* \mathcal{A} *q*

and (*y*, σ) \in set-spmf (*exec-gpv random-permutation* \mathcal{A} σ)

```

and fin: finite (dom  $\sigma$ )
shows card (dom  $\sigma'$ )  $\leq q + \text{card}$  (dom  $\sigma$ )
<proof>

end

end

```

1.12 Reducing games with many adversary guesses to games with single guesses

theory *Guessing-Many-One imports*

CryptHOL.Computational-Model

CryptHOL.GPV-Bisim

begin

locale *guessing-many-one* =

fixes *init* :: ('c-o \times 'c-a \times 's) *spmf*

and *oracle* :: 'c-o \Rightarrow 's \Rightarrow 'call \Rightarrow ('ret \times 's) *spmf*

and *eval* :: 'c-o \Rightarrow 'c-a \Rightarrow 's \Rightarrow 'guess \Rightarrow *bool* *spmf*

begin

type-synonym ('c-a', 'guess', 'call', 'ret') *adversary-single* = 'c-a' \Rightarrow ('guess', 'call', 'ret') *gpv*

definition *game-single* :: ('c-a, 'guess, 'call, 'ret) *adversary-single* \Rightarrow *bool* *spmf*

where

game-single \mathcal{A} = *do* {

(*c-o*, *c-a*, *s*) \leftarrow *init*;

(*guess*, *s'*) \leftarrow *exec-gpv* (*oracle* *c-o*) (\mathcal{A} *c-a*) *s*;

eval *c-o* *c-a* *s'* *guess*

}

definition *advantage-single* :: ('c-a, 'guess, 'call, 'ret) *adversary-single* \Rightarrow *real*

where *advantage-single* \mathcal{A} = *spmf* (*game-single* \mathcal{A}) *True*

type-synonym ('c-a', 'guess', 'call', 'ret') *adversary-many* = 'c-a' \Rightarrow (*unit*, 'call' + 'guess', 'ret' + *unit*) *gpv*

definition *eval-oracle* :: 'c-o \Rightarrow 'c-a \Rightarrow *bool* \times 's \Rightarrow 'guess \Rightarrow (*unit* \times (*bool* \times 's)) *spmf*

where

eval-oracle *c-o* *c-a* = (λ (*b*, *s'*) *guess*. *map-spmf* (λ *b'*. (((), (*b* \vee *b'*, *s'*))) (*eval* *c-o* *c-a* *s'* *guess*))

definition *game-multi* :: ('c-a, 'guess, 'call, 'ret) *adversary-many* \Rightarrow *bool* *spmf*

where

game-multi \mathcal{A} = *do* {

(*c-o*, *c-a*, *s*) \leftarrow *init*;

```

(-, (b, -)) ← exec-gpv
(†(oracle c-o) ⊕O eval-oracle c-o c-a)
(ℳ c-a)
(False, s);
return-spmf b
}

```

definition *advantage-multi* :: ('c-a, 'guess, 'call, 'ret) adversary-many ⇒ real
where *advantage-multi* ℳ = spmf (game-multi ℳ) True

type-synonym 'guess' reduction-state = 'guess' + nat

primrec *process-call* :: 'guess reduction-state ⇒ 'call ⇒ ('ret option × 'guess reduction-state, 'call, 'ret) gpv

where

```

process-call (Inr j) x = do {
  ret ← Pause x Done;
  Done (Some ret, Inr j)
}
| process-call (Inl guess) x = Done (None, Inl guess)

```

primrec *process-guess* :: 'guess reduction-state ⇒ 'guess ⇒ (unit option × 'guess reduction-state, 'call, 'ret) gpv

where

```

process-guess (Inr j) guess = Done (if j > 0 then (Some (), Inr (j - 1)) else (None, Inl guess))
| process-guess (Inl guess) - = Done (None, Inl guess)

```

abbreviation *reduction-oracle* :: 'guess + nat ⇒ 'call + 'guess ⇒ (('ret + unit) option × ('guess + nat), 'call, 'ret) gpv

where *reduction-oracle* ≡ plus-intercept-stop process-call process-guess

definition *reduction* :: nat ⇒ ('c-a, 'guess, 'call, 'ret) adversary-many ⇒ ('c-a, 'guess, 'call, 'ret) adversary-single

where

```

reduction q ℳ c-a = do {
  j-star ← lift-spmf (spmf-of-set {.. $q$ });
  (-, s) ← inline-stop reduction-oracle (ℳ c-a) (Inr j-star);
  Done (projl s)
}

```

lemma *many-single-reduction*:

assumes *bound*: $\bigwedge c-a\ c-o\ s. (c-o, c-a, s) \in \text{set-spmf init} \implies \text{interaction-bounded-by (Not } \circ \text{isl) } (\mathcal{A}\ c-a)\ q$

and *lossless-oracle*: $\bigwedge c-a\ c-o\ s\ s'x. (c-o, c-a, s) \in \text{set-spmf init} \implies \text{lossless-spmf (oracle } c-o\ s'x)$

and *lossless-eval*: $\bigwedge c-a\ c-o\ s\ s'\ \text{guess}. (c-o, c-a, s) \in \text{set-spmf init} \implies \text{lossless-spmf (eval } c-o\ c-a\ s'\ \text{guess)}$

shows *advantage-multi* $\mathcal{A} \leq \text{advantage-single (reduction } q \mathcal{A}) * q$
including *lifting-syntax*
 <proof>

end

end

1.13 Unpredictable function

theory *Unpredictable-Function* **imports**
Guessing-Many-One
begin

locale *upf* =
fixes *key-gen* :: 'key spmf
and *hash* :: 'key \Rightarrow 'x \Rightarrow 'hash
begin

type-synonym ('x', 'hash') *adversary* = (unit, 'x' + ('x' \times 'hash'), 'hash' + unit) *gpv*

definition *oracle-hash* :: 'key \Rightarrow ('x, 'hash, 'x set) *callee*

where

oracle-hash *k* = ($\lambda L y$. *do* {
 let *t* = *hash* *k* *y*;
 let *L* = *insert* *y* *L*;
 return-spmf (*t*, *L*)
 })

definition *oracle-flag* :: 'key \Rightarrow ('x \times 'hash, unit, bool \times 'x set) *callee*

where

oracle-flag = ($\lambda \text{key } (\text{flag}, L)$ (*y*, *t*).
 return-spmf ((, (flag \vee (*t* = (*hash* *key* *y*) \wedge *y* \notin *L*), *L*)))

abbreviation *oracle* :: 'key \Rightarrow ('x + 'x \times 'hash, 'hash + unit, bool \times 'x set) *callee*

where *oracle* *key* \equiv $\dagger(\text{oracle-hash } \text{key}) \oplus_O \text{oracle-flag } \text{key}$

definition *game* :: ('x, 'hash) *adversary* \Rightarrow bool spmf

where

game \mathcal{A} = *do* {
key \leftarrow *key-gen*;
 (-, (*flag'*, *L'*)) \leftarrow *exec-gpv* (*oracle* *key*) \mathcal{A} (*False*, {});
 return-spmf *flag'*
 }

definition *advantage* :: ('x, 'hash) *adversary* \Rightarrow real

where *advantage* \mathcal{A} = spmf (*game* \mathcal{A}) *True*

type-synonym ('x', 'hash') *adversary1* = ('x' \times 'hash', 'x', 'hash') *gpv*

definition *game1* :: ('x, 'hash) adversary1 \Rightarrow bool spmf

where

```
game1  $\mathcal{A}$  = do {  
  key  $\leftarrow$  key-gen;  
  ((m, h), L)  $\leftarrow$  exec-gpv (oracle-hash key)  $\mathcal{A}$  {};  
  return-spmf (h = hash key m  $\wedge$  m  $\notin$  L)  
}
```

definition *advantage1* :: ('x, 'hash) adversary1 \Rightarrow real

where *advantage1* \mathcal{A} = spmf (*game1* \mathcal{A}) True

lemma *advantage-advantage1*:

assumes *bound*: interaction-bounded-by (Not \circ isl) \mathcal{A} *q*

shows *advantage* \mathcal{A} \leq *advantage1* (guessing-many-one.reduction *q* (λ - :: unit. \mathcal{A}) ())

* *q*

\langle proof \rangle

end

end

theory *Security-Spec* imports

Diffie-Hellman

IND-CCA2

IND-CCA2-sym

IND-CPA

IND-CPA-PK

IND-CPA-PK-Single

SUF-CMA

Pseudo-Random-Function

Pseudo-Random-Permutation

Unpredictable-Function

begin

end

2 Cryptographic constructions and their security

theory *Elgamal* imports

CryptHOL.Cyclic-Group-SPMF

CryptHOL.Computational-Model

Diffie-Hellman

IND-CPA-PK-Single

CryptHOL.Negligible

begin

2.1 Elgamal encryption scheme

locale *elgamal-base* =

fixes $\mathcal{G} :: \text{'grp cyclic-group (structure)}$

begin

type-synonym $\text{'grp' pub-key} = \text{'grp'}$

type-synonym $\text{'grp' priv-key} = \text{nat}$

type-synonym $\text{'grp' plain} = \text{'grp'}$

type-synonym $\text{'grp' cipher} = \text{'grp'} \times \text{'grp'}$

definition *key-gen* :: $(\text{'grp pub-key} \times \text{'grp priv-key}) \text{ spmf}$

where

key-gen = do {
 $x \leftarrow \text{sample-uniform (order } \mathcal{G}\text{)}$;
 $\text{return-spmf } (\mathbf{g} [\wedge] x, x)$
}

lemma *key-gen-alt*:

key-gen = $\text{map-spmf } (\lambda x. (\mathbf{g} [\wedge] x, x)) (\text{sample-uniform (order } \mathcal{G}\text{)})$
 $\langle \text{proof} \rangle$

definition *aencrypt* :: $\text{'grp pub-key} \Rightarrow \text{'grp} \Rightarrow \text{'grp cipher spmf}$

where

aencrypt $\alpha \text{ msg} = \text{do } \{$
 $y \leftarrow \text{sample-uniform (order } \mathcal{G}\text{)}$;
 $\text{return-spmf } (\mathbf{g} [\wedge] y, (\alpha [\wedge] y) \otimes \text{msg})$
}

lemma *aencrypt-alt*:

aencrypt $\alpha \text{ msg} = \text{map-spmf } (\lambda y. (\mathbf{g} [\wedge] y, (\alpha [\wedge] y) \otimes \text{msg})) (\text{sample-uniform (order } \mathcal{G}\text{)})$
 $\langle \text{proof} \rangle$

definition *adecrypt* :: $\text{'grp priv-key} \Rightarrow \text{'grp cipher} \Rightarrow \text{'grp option}$

where

adecrypt $x = (\lambda (\beta, \zeta). \text{Some } (\zeta \otimes (\text{inv } (\beta [\wedge] x))))$

abbreviation *valid-plains* :: $\text{'grp} \Rightarrow \text{'grp} \Rightarrow \text{bool}$

where *valid-plains* $\text{msg1 msg2} \equiv \text{msg1} \in \text{carrier } \mathcal{G} \wedge \text{msg2} \in \text{carrier } \mathcal{G}$

sublocale *ind-cpa*: *ind-cpa key-gen aencrypt adecrypt valid-plains* $\langle \text{proof} \rangle$

sublocale *ddh*: *ddh* \mathcal{G} $\langle \text{proof} \rangle$

fun *elgamal-adversary* :: $(\text{'grp pub-key}, \text{'grp plain}, \text{'grp cipher}, \text{'state}) \text{ ind-cpa.adversary}$
 $\Rightarrow \text{'grp ddh.adversary}$

where

elgamal-adversary $(\mathcal{A}1, \mathcal{A}2) \alpha \beta \gamma = \text{TRY do } \{$
 $b \leftarrow \text{coin-spmf}$;
 $((\text{msg1}, \text{msg2}), \sigma) \leftarrow \mathcal{A}1 \alpha$;

— have to check that the attacker actually sends two elements from the group; otherwise


```

flip a coin
- :: unit ← assert-spmf (valid-plains msg1 msg2);
  guess ←  $\mathcal{A}2$  ( $\beta$ ,  $\gamma \otimes$  (if b then msg1 else msg2))  $\sigma$ ;
  return-spmf (guess = b)
} ELSE coin-spmf

end

locale elgamal = elgamal-base + cyclic-group  $\mathcal{G}$ 
begin

theorem advantage-elgamal: ind-cpa.advantage  $\mathcal{A}$  = ddh.advantage (elgamal-adversary
 $\mathcal{A}$ )
including monad-normalisation
⟨proof⟩

end

locale elgamal-asymp =
fixes  $\mathcal{G}$  :: security  $\Rightarrow$  'grp cyclic-group
assumes elgamal:  $\wedge \eta$ . elgamal ( $\mathcal{G}$   $\eta$ )
begin

sublocale elgamal  $\mathcal{G}$   $\eta$  for  $\eta$  ⟨proof⟩

theorem elgamal-secure:
negligible ( $\lambda \eta$ . ind-cpa.advantage  $\eta$  ( $\mathcal{A}$   $\eta$ )) if negligible ( $\lambda \eta$ . ddh.advantage  $\eta$  (elgamal-adversary
 $\eta$  ( $\mathcal{A}$   $\eta$ )))
⟨proof⟩

end

context elgamal-base begin

lemma lossless-key-gen [simp]: lossless-spmf (key-gen)  $\longleftrightarrow$   $0 < \text{order } \mathcal{G}$ 
⟨proof⟩

lemma lossless-aencrypt [simp]:
lossless-spmf (aencrypt key plain)  $\longleftrightarrow$   $0 < \text{order } \mathcal{G}$ 
⟨proof⟩

lemma lossless-elgamal-adversary:
[[ ind-cpa.lossless  $\mathcal{A}$ ;  $0 < \text{order } \mathcal{G}$  ]
 $\implies$  ddh.lossless (elgamal-adversary  $\mathcal{A}$ )
⟨proof⟩

end

end

```

2.2 Hashed Elgamal in the Random Oracle Model

theory *Hashed-Elgamal* **imports**

CryptHOL.GPV-Bisim

CryptHOL.Cyclic-Group-SPMF

CryptHOL.List-Bits

IND-CPA-PK

Diffie-Hellman

begin

type-synonym *bitstring* = *bool list*

locale *hash-oracle* = **fixes** *len* :: *nat* **begin**

type-synonym *'a state* = *'a* \rightarrow *bitstring*

definition *oracle* :: *'a state* \Rightarrow *'a* \Rightarrow (*bitstring* \times *'a state*) *spmf*

where

oracle σ *x* =

(*case* σ *x* of *None* \Rightarrow *do* {
 bs \leftarrow *spmf-of-set* (*nlists UNIV len*);
 return-spmf (*bs*, $\sigma(x \mapsto bs)$)
 } | *Some bs* \Rightarrow *return-spmf* (*bs*, σ))

abbreviation (*input*) *initial* :: *'a state* **where** *initial* \equiv *Map.empty*

inductive invariant :: *'a state* \Rightarrow *bool*

where

invariant: $\llbracket \text{finite } (\text{dom } \sigma); \text{length } ' \text{ran } \sigma \subseteq \{len\} \rrbracket \Longrightarrow \text{invariant } \sigma$

lemma *invariant-initial* [*simp*]: *invariant initial*

<proof>

lemma *invariant-update* [*simp*]: $\llbracket \text{invariant } \sigma; \text{length } bs = len \rrbracket \Longrightarrow \text{invariant } (\sigma(x \mapsto bs))$

<proof>

lemma *invariant* [*intro!*, *simp*]: *callee-invariant oracle invariant*

<proof>

lemma *invariant-in-dom* [*simp*]: *callee-invariant oracle* ($\lambda \sigma. x \in \text{dom } \sigma$)

<proof>

lemma *lossless-oracle* [*simp*]: *lossless-spmf (oracle* σ *x)*

<proof>

lemma *card-dom-state*:

assumes σ' : (*x*, σ') \in *set-spmf (exec-gpv oracle gpv* σ)

and *ibound*: *interaction-any-bounded-by gpv n*

shows *card (dom* σ') $\leq n + \text{card } (\text{dom } \sigma)$

<proof>

end

locale *elgamal-base* =

fixes $\mathcal{G} :: 'grp$ cyclic-group (structure)

and *len-plain* :: nat

begin

sublocale *hash*: hash-oracle *len-plain* *<proof>*

abbreviation *hash* :: 'grp \Rightarrow (bitstring, 'grp, bitstring) gpv

where *hash* $x \equiv$ Pause x Done

type-synonym 'grp' *pub-key* = 'grp'

type-synonym 'grp' *priv-key* = nat

type-synonym *plain* = bitstring

type-synonym 'grp' *cipher* = 'grp' \times bitstring

definition *key-gen* :: ('grp *pub-key* \times 'grp *priv-key*) spmf

where

key-gen = do {
 $x \leftarrow$ sample-uniform (order \mathcal{G});
 return-spmf ($\mathbf{g} [\wedge] x, x$)
}

definition *aencrypt* :: 'grp *pub-key* \Rightarrow *plain* \Rightarrow ('grp *cipher*, 'grp, bitstring) gpv

where

aencrypt α *msg* = do {
 $y \leftarrow$ lift-spmf (sample-uniform (order \mathcal{G}));
 $h \leftarrow$ hash ($\alpha [\wedge] y$);
 Done ($\mathbf{g} [\wedge] y, h [\oplus] \text{msg}$)
}

definition *adecrypt* :: 'grp *priv-key* \Rightarrow 'grp *cipher* \Rightarrow (*plain*, 'grp, bitstring) gpv

where

adecrypt $x = (\lambda(\beta, \zeta). \text{do} \{$
 $h \leftarrow$ hash ($\beta [\wedge] x$);
 Done ($\zeta [\oplus] h$)
})

definition *valid-plains* :: *plain* \Rightarrow *plain* \Rightarrow bool

where *valid-plains* *msg1* *msg2* \iff length *msg1* = *len-plain* \wedge length *msg2* = *len-plain*

lemma *lossless-aencrypt* [simp]: lossless-gpv \mathcal{I} (*aencrypt* α *msg*) \iff $0 <$ order \mathcal{G}

<proof>

lemma *interaction-bounded-by-aencrypt* [interaction-bound, simp]:

interaction-bounded-by ($\lambda-. \text{True}$) (*aencrypt* α *msg*) 1

<proof>

sublocale *ind-cpa*: *ind-cpa-pk lift-spmf key-gen aencrypt adecrypt valid-plains* \langle proof \rangle

sublocale *lcdh*: *lcdh* \mathcal{G} \langle proof \rangle

fun *elgamal-adversary*

$:: ('grp\ pub\ key, plain, 'grp\ cipher, 'grp, bitstring, 'state) ind-cpa.adversary$

$\Rightarrow 'grp\ lcdh.adversary$

where

elgamal-adversary ($\mathcal{A}1, \mathcal{A}2$) $\alpha\ \beta = do$ {

$((msg1, msg2), \sigma), s) \leftarrow exec-gpv\ hash.oracle\ (\mathcal{A}1\ \alpha)\ hash.initial;$

— have to check that the attacker actually sends an element from the group; otherwise

stop early

TRY *do* {

$- :: unit \leftarrow assert-spmf\ (valid-plains\ msg1\ msg2);$

$h' \leftarrow spmf-of-set\ (nlists\ UNIV\ len-plain);$

$(guess, s') \leftarrow exec-gpv\ hash.oracle\ (\mathcal{A}2\ (\beta, h')\ \sigma)\ s;$

$return-spmf\ (dom\ s')$

} *ELSE* *return-spmf* ($dom\ s$)

}

end

locale *elgamal* = *elgamal-base* +

assumes *cyclic-group*: *cyclic-group* \mathcal{G}

begin

interpretation *cyclic-group* \mathcal{G} \langle proof \rangle

lemma *advantage-elgamal*:

includes *lifting-syntax*

assumes *lossless*: *ind-cpa.lossless* \mathcal{A}

shows *ind-cpa.advantage hash.oracle hash.initial* $\mathcal{A} \leq lcdh.advantage\ (elgamal-adversary\ \mathcal{A})$

\langle proof \rangle

including *monad-normalisation* \langle proof \rangle

including *monad-normalisation*

\langle proof \rangle

end

context *elgamal-base* **begin**

lemma *lossless-key-gen* [*simp*]: *lossless-spmf key-gen* $\longleftrightarrow 0 < order\ \mathcal{G}$

\langle proof \rangle

lemma *lossless-elgamal-adversary*:

$\llbracket ind-cpa.lossless\ \mathcal{A}; \wedge \eta. 0 < order\ \mathcal{G} \rrbracket$

$\Longrightarrow lcdh.lossless\ (elgamal-adversary\ \mathcal{A})$

\langle proof \rangle

end

end

2.3 The random-permutation random-function switching lemma

theory *RP-RF imports*

Pseudo-Random-Function

Pseudo-Random-Permutation

CryptHOL.GPV-Bisim

begin

lemma *rp-resample*:

assumes $B \subseteq A \cup C$ $A \cap C = \{\}$ $C \subseteq B$ **and** *finB*: *finite B*

shows $\text{bind-spmf} (\text{spmf-of-set } B) (\lambda x. \text{if } x \in A \text{ then } \text{spmf-of-set } C \text{ else } \text{return-spmf } x) =$
spmf-of-set C

<proof>

locale *rp-rf* =

rp: *random-permutation A* +

rf: *random-function spmf-of-set A*

for *A* :: *'a set*

+

assumes *finite-A*: *finite A*

and *nonempty-A*: $A \neq \{\}$

begin

type-synonym *'a adversary* = (*bool*, *'a*, *'a*) *gpv*

definition *game* :: *bool* \Rightarrow *'a adversary* \Rightarrow *bool spmf* **where**

game b $\mathcal{A} = \text{run-gpv} (\text{if } b \text{ then } \text{rp.random-permutation} \text{ else } \text{rf.random-oracle}) \mathcal{A}$
Map.empty

abbreviation *prp-game* :: *'a adversary* \Rightarrow *bool spmf* **where** *prp-game* \equiv *game True*

abbreviation *prf-game* :: *'a adversary* \Rightarrow *bool spmf* **where** *prf-game* \equiv *game False*

definition *advantage* :: *'a adversary* \Rightarrow *real* **where**

advantage $\mathcal{A} = |\text{spmf} (\text{prp-game } \mathcal{A}) \text{ True} - \text{spmf} (\text{prf-game } \mathcal{A}) \text{ True}|$

lemma *advantage-nonneg*: $0 \leq \text{advantage } \mathcal{A}$ *<proof>*

lemma *advantage-le-1*: $\text{advantage } \mathcal{A} \leq 1$

<proof>

context includes *\mathcal{I}.lifting* **begin**

lift-definition *\mathcal{I}* :: (*'a*, *'a*) *\mathcal{I}* **is** $(\lambda x. \text{if } x \in A \text{ then } A \text{ else } \{\})$ *<proof>*

lemma *outs-\mathcal{I}-\mathcal{I}* [*simp*]: *outs-\mathcal{I} \mathcal{I} = A* *<proof>*

lemma *responses-\mathcal{I}-\mathcal{I}* [*simp*]: *responses-\mathcal{I} \mathcal{I} x = (if x \in A then A else \{\})* *<proof>*

```

lifting-update  $\mathcal{I}$ .lifting
lifting-forget  $\mathcal{I}$ .lifting
end

```

```

lemma rp-rf:
  assumes bound: interaction-any-bounded-by  $\mathcal{A}$   $q$ 
    and lossless: lossless-gpv  $\mathcal{I}$   $\mathcal{A}$ 
    and WT:  $\mathcal{I} \vdash g$   $\mathcal{A} \checkmark$ 
  shows advantage  $\mathcal{A} \leq q * q / \text{card } A$ 
  including lifting-syntax
   $\langle \text{proof} \rangle$ 

```

```

end

```

```

end

```

2.4 Extending the input length of a PRF using a universal hash function

This example is taken from [19, §4.2].

```

theory PRF-UHF imports
  CryptHOL.GPV-Bisim
  Pseudo-Random-Function
begin

```

```

locale hash =
  fixes seed-gen :: 'seed spmf
  and hash :: 'seed  $\Rightarrow$  'domain  $\Rightarrow$  'range
begin

```

```

definition game-hash :: 'domain  $\Rightarrow$  'domain  $\Rightarrow$  bool spmf
where
  game-hash  $w w' = \text{do}$  {
    seed  $\leftarrow$  seed-gen;
    return-spmf (hash seed  $w = \text{hash seed } w' \wedge w \neq w'$ )
  }

```

```

definition game-hash-set :: 'domain set  $\Rightarrow$  bool spmf
where
  game-hash-set  $W = \text{do}$  {
    seed  $\leftarrow$  seed-gen;
    return-spmf ( $\neg \text{inj-on } (\text{hash } \text{seed}) W$ )
  }

```

```

definition  $\varepsilon\text{-uh}$  :: real
where  $\varepsilon\text{-uh} = (\text{SUP } w w'. \text{spmf } (\text{game-hash } w w') \text{ True})$ 

```

```

lemma  $\varepsilon\text{-uh-nonneg}$  :  $\varepsilon\text{-uh} \geq 0$ 
   $\langle \text{proof} \rangle$ 

```

lemma *hash-ineq-card*:
assumes *finite W*
shows *spmf (game-hash-set W) True* $\leq \epsilon$ -uh * *card W* * *card W*
 ⟨*proof*⟩

end

locale *prf-hash* =
fixes *f* :: 'key \Rightarrow 'α \Rightarrow 'γ
and *h* :: 'seed \Rightarrow 'β \Rightarrow 'α
and *key-gen* :: 'key *spmf*
and *seed-gen* :: 'seed *spmf*
and *range-f* :: 'γ *set*
assumes *lossless-seed-gen*: *lossless-spmf seed-gen*
and *range-f-finite*: *finite range-f*
and *range-f-nonempty*: *range-f* $\neq \{\}$
begin

definition *rand* :: 'γ *spmf*
where *rand* = *spmf-of-set range-f*

lemma *lossless-rand [simp]*: *lossless-spmf rand*
 ⟨*proof*⟩

definition *key-seed-gen* :: ('key * 'seed) *spmf*
where
key-seed-gen = do {
 k \leftarrow *key-gen*;
 s :: 'seed \leftarrow *seed-gen*;
 return-*spmf* (*k*, *s*)
}

interpretation *prf*: *prf key-gen f rand* ⟨*proof*⟩

interpretation *hash*: *hash seed-gen h* ⟨*proof*⟩

fun *f'* :: 'key \times 'seed \Rightarrow 'β \Rightarrow 'γ
where *f'* (*key*, *seed*) *x* = *f key (h seed x)*

interpretation *prf'*: *prf key-seed-gen f' rand* ⟨*proof*⟩

definition *reduction-oracle* :: 'seed \Rightarrow unit \Rightarrow 'β \Rightarrow ('γ \times unit, 'α, 'γ) *gpv*
where *reduction-oracle seed x b* = *Pause (h seed b) (λx. Done (x, ()))*

definition *prf'-reduction* :: ('β, 'γ) *prf'.adversary* \Rightarrow ('α, 'γ) *prf.adversary*
where

prf'-reduction \mathcal{A} = do {
 seed \leftarrow *lift-spmf seed-gen*;
 (*b*, σ) \leftarrow *inline (reduction-oracle seed) \mathcal{A} ()*;

```

    Done b
  }

theorem prf-prf'-advantage:
  assumes prf'.lossless  $\mathcal{A}$ 
  and bounded: prf'.ibounded-by  $\mathcal{A}$   $q$ 
  shows prf'.advantage  $\mathcal{A} \leq$  prf.advantage (prf'-reduction  $\mathcal{A}$ ) + hash. $\epsilon$ -uh *  $q * q$ 
  including lifting-syntax
   $\langle$ proof $\rangle$  including monad-normalisation
   $\langle$ proof $\rangle$ 

end

end

```

2.5 IND-CPA from PRF

theory PRF-IND-CPA **imports**

```

  CryptHOL.GPV-Bisim
  CryptHOL.List-Bits
  Pseudo-Random-Function
  IND-CPA

```

begin

Formalises the construction from [16].

declare $[[\text{simproc del: let-simp}]]$

```

type-synonym key = bool list
type-synonym plain = bool list
type-synonym cipher = bool list * bool list

```

locale otp =

fixes $f :: \text{key} \Rightarrow \text{bool list} \Rightarrow \text{bool list}$

and $len :: \text{nat}$

assumes $\text{length-f: } \bigwedge xs \ ys. \llbracket \text{length } xs = len; \text{length } ys = len \rrbracket \implies \text{length } (f \ xs \ ys) = len$

begin

definition $\text{key-gen} :: \text{bool list} \text{ spmf}$

where $\text{key-gen} = \text{spmf-of-set } (nlists \ UNIV \ len)$

definition $\text{valid-plain} :: \text{plain} \Rightarrow \text{bool}$

where $\text{valid-plain } plain \longleftrightarrow \text{length } plain = len$

definition $\text{encrypt} :: \text{key} \Rightarrow \text{plain} \Rightarrow \text{cipher} \text{ spmf}$

where

```

  encrypt key plain = do {
     $r \leftarrow \text{spmf-of-set } (nlists \ UNIV \ len)$ ;
    return-spmf (r, xor-list plain (f key r))
  }

```


fun *decrypt* :: *key* ⇒ *cipher* ⇒ *plain option*
where *decrypt key (r, c) = Some (xor-list (f key r) c)*

lemma *encrypt-decrypt-correct*:
 [[*length key = len; length plain = len*]]
 ⇒ *encrypt key plain* >>= (λ *cipher. return-spmf (decrypt key cipher)*) = *return-spmf*
 (*Some plain*)
 ⟨*proof*⟩

interpretation *ind-cpa*: *ind-cpa key-gen encrypt decrypt valid-plain* ⟨*proof*⟩
interpretation *prf*: *prf key-gen f spmf-of-set (nlists UNIV len)* ⟨*proof*⟩

definition *prf-encrypt-oracle* :: *unit* ⇒ *plain* ⇒ (*cipher* × *unit, plain, plain*) *gpv*
where
prf-encrypt-oracle x plain = do {
 r ← *lift-spmf (spmf-of-set (nlists UNIV len))*;
 Pause r (λpad. Done ((r, xor-list plain pad), ()))
 }

lemma *interaction-bounded-by-prf-encrypt-oracle* [*interaction-bound*]:
interaction-any-bounded-by (prf-encrypt-oracle σ plain) 1
 ⟨*proof*⟩

lemma *lossless-prf-encrypt-oracle* [*simp*]: *lossless-gpv ℒ-top (prf-encrypt-oracle s x)*
 ⟨*proof*⟩

definition *prf-adversary* :: (*plain, cipher, 'state*) *ind-cpa.adversary* ⇒ (*plain, plain*) *prf.adversary*
where
prf-adversary ℒ = do {
 let (ℒ1, ℒ2) = ℒ;
 (*((p1, p2), σ), n*) ← *inline prf-encrypt-oracle ℒ1 ()*;
 if valid-plain p1 ∧ valid-plain p2 then do {
 b ← *lift-spmf coin-spmf*;
 let pb = (if b then p1 else p2);
 r ← *lift-spmf (spmf-of-set (nlists UNIV len))*;
 pad ← *Pause r Done*;
 let c = (r, xor-list pb pad);
 (*b', -*) ← *inline prf-encrypt-oracle (ℒ2 c σ) n*;
 Done (b' = b)
 } *else lift-spmf coin-spmf*
 }

theorem *prf-encrypt-advantage*:
assumes *ind-cpa.ibounded-by ℒ q*
and *lossless-gpv ℒ-full (fst ℒ)*
and \bigwedge *cipher σ. lossless-gpv ℒ-full (snd ℒ cipher σ)*
shows *ind-cpa.advantage ℒ* ≤ *prf.advantage (prf-adversary ℒ) + q / 2 ^ len*
 ⟨*proof*⟩

```

including monad-normalisation ⟨proof⟩ including monad-normalisation
⟨proof⟩
including monad-normalisation ⟨proof⟩

lemma interaction-bounded-prf-adversary:
  fixes q :: nat
  assumes ind-cpa.ibounded-by  $\mathcal{A}$  q
  shows prf.ibounded-by (prf-adversary  $\mathcal{A}$ ) (1 + q)
  ⟨proof⟩

lemma lossless-prf-adversary: ind-cpa.lossless  $\mathcal{A}$   $\implies$  prf.lossless (prf-adversary  $\mathcal{A}$ )
  ⟨proof⟩

end

locale otp- $\eta$  =
  fixes f :: security  $\Rightarrow$  key  $\Rightarrow$  bool list  $\Rightarrow$  bool list
  and len :: security  $\Rightarrow$  nat
  assumes length-f:  $\bigwedge \eta$  xs ys.  $\llbracket$  length xs = len  $\eta$ ; length ys = len  $\eta$   $\rrbracket \implies$  length (f  $\eta$  xs
  ys) = len  $\eta$ 
  and negligible-len [negligible-intros]: negligible ( $\lambda \eta. 1 / 2 ^ (\text{len } \eta)$ )
  begin

  interpretation otp f  $\eta$  len  $\eta$  for  $\eta$  ⟨proof⟩
  interpretation ind-cpa: ind-cpa key-gen  $\eta$  encrypt  $\eta$  decrypt  $\eta$  valid-plain  $\eta$  for  $\eta$  ⟨proof⟩
  interpretation prf: prf key-gen  $\eta$  f  $\eta$  spmf-of-set (nlists UNIV (len  $\eta$ )) for  $\eta$  ⟨proof⟩

  lemma prf-encrypt-secure-for:
    assumes [negligible-intros]: negligible ( $\lambda \eta. \text{prf.} \text{advantage } \eta$  (prf-adversary  $\eta$  ( $\mathcal{A}$   $\eta$ )))
    and q:  $\bigwedge \eta. \text{ind-cpa.} \text{ibounded-by } (\mathcal{A} \ \eta)$  (q  $\eta$ ) and [negligible-intros]: polynomial q
    and lossless:  $\bigwedge \eta. \text{ind-cpa.} \text{lossless } (\mathcal{A} \ \eta)$ 
    shows negligible ( $\lambda \eta. \text{ind-cpa.} \text{advantage } \eta$  ( $\mathcal{A} \ \eta$ ))
    ⟨proof⟩

  end

end

```

2.6 IND-CCA from a PRF and an unpredictable function

```

theory PRF-UPF-IND-CCA
imports
  Pseudo-Random-Function
  CryptHOL.List-Bits
  Unpredictable-Function
  IND-CCA2-sym
  CryptHOL.Negligible
begin

```

Formalisation of Shoup's construction of an IND-CCA secure cipher from a PRF

and an unpredictable function [19, §7].

type-synonym *bitstring* = *bool list*

locale *simple-cipher* =

PRF: *prf prf-key-gen prf-fun spmf-of-set (nlists UNIV prf-clen) +*

UPF: *upf upf-key-gen upf-fun*

for *prf-key-gen* :: '*prf-key spmf*

and *prf-fun* :: '*prf-key ⇒ bitstring ⇒ bitstring*

and *prf-domain* :: *bitstring set*

and *prf-range* :: *bitstring set*

and *prf-dlen* :: *nat*

and *prf-clen* :: *nat*

and *upf-key-gen* :: '*upf-key spmf*

and *upf-fun* :: '*upf-key ⇒ bitstring ⇒ 'hash*

+

assumes *prf-domain-finite*: *finite prf-domain*

assumes *prf-domain-nonempty*: *prf-domain ≠ {}*

assumes *prf-domain-length*: *x ∈ prf-domain ⇒ length x = prf-dlen*

assumes *prf-codomain-length*:

$\llbracket \text{key-prf} \in \text{set-spmf } \text{prf-key-gen}; m \in \text{prf-domain} \rrbracket \implies \text{length } (\text{prf-fun } \text{key-prf } m) = \text{prf-clen}$

assumes *prf-key-gen-lossless*: *lossless-spmf prf-key-gen*

assumes *upf-key-gen-lossless*: *lossless-spmf upf-key-gen*

begin

type-synonym '*hash*' *cipher-text* = *bitstring × bitstring × 'hash*

definition *key-gen* :: (*prf-key × upf-key*) *spmf where*

key-gen = *do* {

k-prf ← *prf-key-gen*;

k-upf :: '*upf-key* ← *upf-key-gen*;

return-spmf (*k-prf*, *k-upf*)

}

lemma *lossless-key-gen* [*simp*]: *lossless-spmf key-gen*

<proof>

fun *encrypt* :: (*prf-key × upf-key*) ⇒ *bitstring ⇒ 'hash cipher-text spmf*

where

encrypt (*k-prf*, *k-upf*) *m* = *do* {

x ← *spmf-of-set prf-domain*;

let c = *prf-fun k-prf x* [⊕] *m*;

let t = *upf-fun k-upf (x @ c)*;

return-spmf ((*x*, *c*, *t*))

}

lemma *lossless-encrypt* [*simp*]: *lossless-spmf (encrypt k m)*

<proof>

```

fun decrypt :: ('prf-key × 'upf-key) ⇒ 'hash cipher-text ⇒ bitstring option
where
  decrypt (k-prf, k-upf) (x, c, t) = (
    if upf-fun k-upf (x @ c) = t ∧ length x = prf-dlen then
      Some (prf-fun k-prf x [⊕] c)
    else
      None
  )

```

```

lemma cipher-correct:
  [[ k ∈ set-spmf key-gen; length m = prf-clen ]]
  ⇒ encrypt k m ≫= (λc. return-spmf (decrypt k c)) = return-spmf (Some m)
  ⟨proof⟩

```

```

declare encrypt.simps[simp del]

```

```

sublocale ind-cca: ind-cca key-gen encrypt decrypt λm. length m = prf-clen ⟨proof⟩
interpretation ind-cca': ind-cca key-gen encrypt λ - -. None λm. length m = prf-clen
  ⟨proof⟩

```

```

definition intercept-upf-enc
  :: 'prf-key ⇒ bool ⇒ 'hash cipher-text set × 'hash cipher-text set ⇒ bitstring × bitstring
  ⇒ ('hash cipher-text option × ('hash cipher-text set × 'hash cipher-text set),
    bitstring + (bitstring × 'hash), 'hash + unit) gpv
where
  intercept-upf-enc k b = (λ(L, D) (m1, m0).
    (case (length m1 = prf-clen ∧ length m0 = prf-clen) of
      False ⇒ Done (None, L, D)
    | True ⇒ do {
      x ← lift-spmf (spmf-of-set prf-domain);
      let c = prf-fun k x [⊕] (if b then m1 else m0);
      t ← Pause (Inl (x @ c)) Done;
      Done ((Some (x, c, proj1 t)), (insert (x, c, proj1 t) L, D))
    })))

```

```

definition intercept-upf-dec
  :: 'hash cipher-text set × 'hash cipher-text set ⇒ 'hash cipher-text
  ⇒ (bitstring option × ('hash cipher-text set × 'hash cipher-text set),
    bitstring + (bitstring × 'hash), 'hash + unit) gpv
where
  intercept-upf-dec = (λ(L, D) (x, c, t).
    if (x, c, t) ∈ L ∨ length x ≠ prf-dlen then Done (None, (L, D)) else do {
      Pause (Inr (x @ c, t)) Done;
      Done (None, (L, insert (x, c, t) D))
    })

```

```

definition intercept-upf ::
  'prf-key ⇒ bool ⇒ 'hash cipher-text set × 'hash cipher-text set ⇒ bitstring × bitstring
  + 'hash cipher-text

```

$\Rightarrow (('hash\ cipher\text{-}text\ option + bitstring\ option) \times ('hash\ cipher\text{-}text\ set \times 'hash\ cipher\text{-}text\ set),$

$bitstring + (bitstring \times 'hash), 'hash + unit) gpv$

where

$intercept\text{-}upf\ k\ b = plus\text{-}intercept\ (intercept\text{-}upf\text{-}enc\ k\ b)\ intercept\text{-}upf\text{-}dec$

lemma $intercept\text{-}upf\text{-}simps$ [simp]:

$intercept\text{-}upf\ k\ b\ (L, D)\ (Inr\ (x, c, t)) =$

$(if\ (x, c, t) \in L \vee length\ x \neq prf\text{-}dlen\ then\ Done\ (Inr\ None, (L, D))\ else\ do\ \{$

$\ \ Pause\ (Inr\ (x\ @\ c, t))\ Done;$

$\ \ Done\ (Inr\ None, (L, insert\ (x, c, t)\ D))$

$\ \ \})$

$intercept\text{-}upf\ k\ b\ (L, D)\ (Inl\ (m1, m0)) =$

$(case\ (length\ m1 = prf\text{-}clen \wedge length\ m0 = prf\text{-}clen)\ of$

$\ \ False \Rightarrow Done\ (Inl\ None, L, D)$

$\ \ | True \Rightarrow do\ \{$

$\ \ \ \ x \leftarrow lift\text{-}spmf\ (spmf\text{-}of\text{-}set\ prf\text{-}domain);$

$\ \ \ \ let\ c = prf\text{-}fun\ k\ x\ [\oplus]\ (if\ b\ then\ m1\ else\ m0);$

$\ \ \ \ t \leftarrow Pause\ (Inl\ (x\ @\ c))\ Done;$

$\ \ \ \ Done\ (Inl\ (Some\ (x, c, proj1\ t)), (insert\ (x, c, proj1\ t)\ L, D))$

$\ \ \ \ \})$

$\ \ \langle proof \rangle$

lemma $interaction\text{-}bounded\text{-}by\text{-}upf\text{-}enc\text{-}Inr$ [interaction-bound]:

$interaction\text{-}bounded\text{-}by\ (Not \circ isl)\ (intercept\text{-}upf\text{-}enc\ k\ b\ LD\ mm)\ 0$

$\langle proof \rangle$

lemma $interaction\text{-}bounded\text{-}by\text{-}upf\text{-}dec\text{-}Inr$ [interaction-bound]:

$interaction\text{-}bounded\text{-}by\ (Not \circ isl)\ (intercept\text{-}upf\text{-}dec\ LD\ c)\ 1$

$\langle proof \rangle$

lemma $interaction\text{-}bounded\text{-}by\text{-}intercept\text{-}upf\text{-}Inr$ [interaction-bound]:

$interaction\text{-}bounded\text{-}by\ (Not \circ isl)\ (intercept\text{-}upf\ k\ b\ LD\ x)\ 1$

$\langle proof \rangle$

lemma $interaction\text{-}bounded\text{-}by\text{-}intercept\text{-}upf\text{-}Inl$ [interaction-bound]:

$isl\ x \implies interaction\text{-}bounded\text{-}by\ (Not \circ isl)\ (intercept\text{-}upf\ k\ b\ LD\ x)\ 0$

$\langle proof \rangle$

lemma $lossless\text{-}intercept\text{-}upf\text{-}enc$ [simp]: $lossless\text{-}gpv\ (\mathcal{I}\text{-}full \oplus_{\mathcal{I}} \mathcal{I}\text{-}full)\ (intercept\text{-}upf\text{-}enc\ k\ b\ LD\ mm)$

$\langle proof \rangle$

lemma $lossless\text{-}intercept\text{-}upf\text{-}dec$ [simp]: $lossless\text{-}gpv\ (\mathcal{I}\text{-}full \oplus_{\mathcal{I}} \mathcal{I}\text{-}full)\ (intercept\text{-}upf\text{-}dec\ LD\ mm)$

$\langle proof \rangle$

lemma $lossless\text{-}intercept\text{-}upf$ [simp]: $lossless\text{-}gpv\ (\mathcal{I}\text{-}full \oplus_{\mathcal{I}} \mathcal{I}\text{-}full)\ (intercept\text{-}upf\ k\ b)$

LD x
 ⟨proof⟩

lemma *results-gpv-intercept-upf* [simp]: *results-gpv* (\mathcal{I} -full $\oplus_{\mathcal{I}}$ \mathcal{I} -full) (*intercept-upf* *k b LD x*) \subseteq *responses-* \mathcal{I} (\mathcal{I} -full $\oplus_{\mathcal{I}}$ \mathcal{I} -full) *x* \times *UNIV*
 ⟨proof⟩

definition *reduction-upf* :: (*bitstring*, *'hash cipher-text*) *ind-cca.adversary*
 \Rightarrow (*bitstring*, *'hash*) *UPF.adversary*
where *reduction-upf* $\mathcal{A} = \text{do}$ {
 k \leftarrow *lift-spmf prf-key-gen*;
 b \leftarrow *lift-spmf coin-spmf*;
 (\cdot , (*L*, *D*)) \leftarrow *inline* (*intercept-upf k b*) \mathcal{A} ({}, {});
 Done () }

lemma *lossless-reduction-upf* [simp]:
lossless-gpv (\mathcal{I} -full $\oplus_{\mathcal{I}}$ \mathcal{I} -full) $\mathcal{A} \Longrightarrow$ *lossless-gpv* (\mathcal{I} -full $\oplus_{\mathcal{I}}$ \mathcal{I} -full) (*reduction-upf* \mathcal{A})
 ⟨proof⟩

context includes *lifting-syntax begin*

lemma *round-1*:
assumes *lossless-gpv* (\mathcal{I} -full $\oplus_{\mathcal{I}}$ \mathcal{I} -full) \mathcal{A}
shows $| \text{spmf} (\text{ind-cca.game } \mathcal{A}) \text{ True} - \text{spmf} (\text{ind-cca'.game } \mathcal{A}) \text{ True} | \leq \text{UPF.} \text{advantage}$
 (*reduction-upf* \mathcal{A})
 ⟨proof⟩ **including** *monad-normalisation*
 ⟨proof⟩

definition *oracle-encrypt2* ::
 (*'prf-key* \times *'upf-key*) \Rightarrow *bool* \Rightarrow (*bitstring*, *bitstring*) *PRF.dict* \Rightarrow *bitstring* \times *bitstring*
 \Rightarrow (*'hash cipher-text option* \times (*bitstring*, *bitstring*) *PRF.dict*) *spmf*
where
oracle-encrypt2 = (λ (*k-prf*, *k-upf*) *b D* (*msg1*, *msg0*). (*case* (*length msg1* = *prf-clen* \wedge *length msg0* = *prf-clen*) *of*
 False \Rightarrow *return-spmf* (*None*, *D*)
 | *True* \Rightarrow *do* {
 x \leftarrow *spmf-of-set prf-domain*;
 P \leftarrow *spmf-of-set* (*nlists UNIV prf-clen*);
 let *p* = (*case D x of Some r* \Rightarrow *r* | *None* \Rightarrow *P*);
 let *c* = *p* [\oplus] (*if b then msg1 else msg0*);
 let *t* = *upf-fun k-upf* (*x @ c*);
 return-spmf (*Some* (*x*, *c*, *t*), *D*(*x* \mapsto *p*))
 })

definition *oracle-decrypt2*:: (*'prf-key* \times *'upf-key*) \Rightarrow (*'hash cipher-text*, *bitstring option*, *'state*) *callee*
where *oracle-decrypt2* = (λ *key D cipher*. *return-spmf* (*None*, *D*))

lemma *lossless-oracle-decrypt2* [simp]: *lossless-spmf (oracle-decrypt2 k Dbad c)*
 ⟨proof⟩

lemma *callee-invariant-oracle-decrypt2* [simp]: *callee-invariant (oracle-decrypt2 key) fst*
 ⟨proof⟩

lemma *oracle-decrypt2-parametric* [transfer-rule]:
 (*rel-prod P U* \implies *S* \implies *rel-prod (=) (rel-prod (=) H)* \implies *rel-spmf (rel-prod (=) S)*)
oracle-decrypt2 oracle-decrypt2
 ⟨proof⟩

definition *game2* :: (*bitstring*, *'hash cipher-text*) *ind-cca.adversary* \Rightarrow *bool* *spmf*
where
game2 $\mathcal{A} \equiv$ *do* {
key \leftarrow *key-gen*;
b \leftarrow *coin-spmf*;
 (*b'*, *D*) \leftarrow *exec-gpv*
 (*oracle-encrypt2 key b* \oplus_O *oracle-decrypt2 key*) \mathcal{A} *Map-empty*;
return-spmf (b = b')
 }

fun *intercept-prf* ::
'upf-key \Rightarrow *bool* \Rightarrow *unit* \Rightarrow (*bitstring* \times *bitstring*) + *'hash cipher-text*
 \Rightarrow ((*'hash cipher-text option* + *bitstring option*) \times *unit*, *bitstring*, *bitstring*) *gpv*
where
intercept-prf - - (*Inr* -) = *Done (Inr None, ())*
 | *intercept-prf* *k b* - (*Inl (m1, m0)*) = (*case (length m1) = prf-clen* \wedge (*length m0) = prf-clen*
 of
 False \Rightarrow *Done (Inl None, ())*
 | *True* \Rightarrow *do* {
 x \leftarrow *lift-spmf (spmf-of-set prf-domain)*;
 p \leftarrow *Pause x Done*;
 let c = p [\oplus] (*if b then m1 else m0*);
 let t = upf-fun k (x @ c);
 Done (Inl (Some (x, c, t)), ())
 })

definition *reduction-prf*
 :: (*bitstring*, *'hash cipher-text*) *ind-cca.adversary* \Rightarrow (*bitstring*, *bitstring*) *PRF.adversary*
where
reduction-prf $\mathcal{A} =$ *do* {
k \leftarrow *lift-spmf upf-key-gen*;
b \leftarrow *lift-spmf coin-spmf*;
 (*b'*, -) \leftarrow *inline (intercept-prf k b)* \mathcal{A} ();
Done (b' = b)
 }

lemma round-2: $| \text{spmf} (\text{ind-cca}'.\text{game } \mathcal{A}) \text{ True} - \text{spmf} (\text{game2 } \mathcal{A}) \text{ True} | = \text{PRF.} \text{advantage}$
(reduction-prf \mathcal{A})
<proof>

definition oracle-encrypt3 ::

$(\text{'prf-key} \times \text{'upf-key}) \Rightarrow \text{bool} \Rightarrow (\text{bool} \times (\text{bitstring}, \text{bitstring}) \text{PRF.dict}) \Rightarrow$
 $\text{bitstring} \times \text{bitstring} \Rightarrow (\text{'hash cipher-text option} \times (\text{bool} \times (\text{bitstring}, \text{bitstring})$
 $\text{PRF.dict})) \text{spmf}$

where

$\text{oracle-encrypt3} = (\lambda (k\text{-prf}, k\text{-upf}) b (\text{bad}, D) (\text{msg1}, \text{msg0}).$
 $(\text{case } (\text{length msg1} = \text{prf-clen} \wedge \text{length msg0} = \text{prf-clen}) \text{ of}$
 $\text{False} \Rightarrow \text{return-spmf } (\text{None}, (\text{bad}, D))$
 $| \text{True} \Rightarrow \text{do } \{$
 $\quad x \leftarrow \text{spmf-of-set } \text{prf-domain};$
 $\quad P \leftarrow \text{spmf-of-set } (\text{nlists UNIV prf-clen});$
 $\quad \text{let } (p, F) = (\text{case } D \ x \ \text{of } \text{Some } r \Rightarrow (P, \text{True}) \ | \ \text{None} \Rightarrow (P, \text{False}));$
 $\quad \text{let } c = p \ [\oplus] \ (\text{if } b \ \text{then } \text{msg1} \ \text{else } \text{msg0});$
 $\quad \text{let } t = \text{upf-fun } k\text{-upf } (x \ @ \ c);$
 $\quad \text{return-spmf } (\text{Some } (x, c, t), (\text{bad} \vee F, D(x \mapsto p)))$
 $\quad \})$

lemma lossless-oracle-encrypt3 [simp]:
 $\text{lossless-spmf } (\text{oracle-encrypt3 } k \ b \ D \ m10)$
<proof>

lemma callee-invariant-oracle-encrypt3 [simp]: *callee-invariant (oracle-encrypt3 key b)*
fst
<proof>

definition game3 :: $(\text{bitstring}, \text{'hash cipher-text}) \text{ind-cca.adversary} \Rightarrow (\text{bool} \times \text{bool}) \text{spmf}$

where

$\text{game3 } \mathcal{A} \equiv \text{do } \{$
 $\quad \text{key} \leftarrow \text{key-gen};$
 $\quad b \leftarrow \text{coin-spmf};$
 $\quad (b', (\text{bad}, D)) \leftarrow \text{exec-gpv } (\text{oracle-encrypt3 } \text{key } b \ \oplus_O \ \text{oracle-decrypt2 } \text{key}) \ \mathcal{A} \ (\text{False},$
 $\text{Map-empty});$
 $\quad \text{return-spmf } (b = b', \text{bad})$
 $\quad \}$

lemma round-3:

assumes $\text{lossless-gpv } (\mathcal{I}\text{-full} \oplus_{\mathcal{I}} \mathcal{I}\text{-full}) \ \mathcal{A}$
shows $|\text{measure } (\text{measure-spmf } (\text{game3 } \mathcal{A})) \{(b, \text{bad}). b\} - \text{spmf } (\text{game2 } \mathcal{A}) \text{ True}|$
 $\leq \text{measure } (\text{measure-spmf } (\text{game3 } \mathcal{A})) \{(b, \text{bad}). \text{bad}\}$
<proof>

lemma round-4:

assumes $\text{lossless-gpv } (\mathcal{I}\text{-full} \oplus_{\mathcal{I}} \mathcal{I}\text{-full}) \ \mathcal{A}$

shows $\text{map-spmf fst (game3 } \mathcal{A}) = \text{coin-spmf}$
 ⟨proof⟩ **including** *monad-normalisation*
 ⟨proof⟩

lemma *game3-bad*:

assumes *interaction-bounded-by isl* \mathcal{A} q
shows $\text{measure (measure-spmf (game3 } \mathcal{A})) \{(b, \text{bad}). \text{bad}\} \leq q / \text{card prf-domain} * q$
 ⟨proof⟩

theorem *security*:

assumes *lossless*: *lossless-gpv* $(\mathcal{I}\text{-full} \oplus_{\mathcal{I}} \mathcal{I}\text{-full}) \mathcal{A}$
and *bound*: *interaction-bounded-by isl* \mathcal{A} q
shows $\text{ind-cca.advantage } \mathcal{A} \leq$
 $\text{PRF.advantage (reduction-prf } \mathcal{A}) + \text{UPF.advantage (reduction-upf } \mathcal{A}) +$
 $\text{real } q / \text{real (card prf-domain)} * \text{real } q$ (**is** ?LHS ≤ -)
 ⟨proof⟩

theorem *security1*:

assumes *lossless*: *lossless-gpv* $(\mathcal{I}\text{-full} \oplus_{\mathcal{I}} \mathcal{I}\text{-full}) \mathcal{A}$
assumes q : *interaction-bounded-by isl* \mathcal{A} q
and q' : *interaction-bounded-by* $(\text{Not} \circ \text{isl}) \mathcal{A}$ q'
shows $\text{ind-cca.advantage } \mathcal{A} \leq$
 $\text{PRF.advantage (reduction-prf } \mathcal{A}) +$
 $\text{UPF.advantage1 (guessing-many-one.reduction } q' (\lambda\text{-. reduction-upf } \mathcal{A}) ()) * q' +$
 $\text{real } q * \text{real } q / \text{real (card prf-domain)}$
 ⟨proof⟩

end

end

locale *simple-cipher'* =

fixes *prf-key-gen* :: *security* \Rightarrow 'prf-key spmf
and *prf-fun* :: *security* \Rightarrow 'prf-key \Rightarrow bitstring \Rightarrow bitstring
and *prf-domain* :: *security* \Rightarrow bitstring set
and *prf-range* :: *security* \Rightarrow bitstring set
and *prf-dlen* :: *security* \Rightarrow nat
and *prf-clen* :: *security* \Rightarrow nat
and *upf-key-gen* :: *security* \Rightarrow 'upf-key spmf
and *upf-fun* :: *security* \Rightarrow 'upf-key \Rightarrow bitstring \Rightarrow 'hash
assumes *simple-cipher*: $\bigwedge \eta. \text{simple-cipher (prf-key-gen } \eta) (\text{prf-fun } \eta) (\text{prf-domain } \eta)$
 $(\text{prf-dlen } \eta) (\text{prf-clen } \eta) (\text{upf-key-gen } \eta)$
begin

sublocale *simple-cipher*

prf-key-gen η *prf-fun* η *prf-domain* η *prf-range* η *prf-dlen* η *prf-clen* η *upf-key-gen* η
upf-fun η
for η

<proof>

theorem *security-asymptotic*:

fixes $q\ q' :: \text{security} \Rightarrow \text{nat}$

assumes *lossless*: $\bigwedge \eta. \text{lossless-gpv } (\mathcal{I}\text{-full} \oplus_{\mathcal{I}} \mathcal{I}\text{-full}) (\mathcal{A} \ \eta)$

and *bound*: $\bigwedge \eta. \text{interaction-bounded-by } \text{isl} (\mathcal{A} \ \eta) (q \ \eta)$

and *bound'*: $\bigwedge \eta. \text{interaction-bounded-by } (\text{Not} \circ \text{isl}) (\mathcal{A} \ \eta) (q' \ \eta)$

and [*negligible-intros*]:

polynomial q' *polynomial* q

negligible $(\lambda \eta. \text{PRF.} \text{advantage } \eta (\text{reduction-prf } \eta (\mathcal{A} \ \eta)))$

negligible $(\lambda \eta. \text{UPF.} \text{advantage1 } \eta (\text{guessing-many-one.reduction } (q' \ \eta) (\lambda \cdot. \text{reduction-upf } \eta (\mathcal{A} \ \eta)) ()))$

negligible $(\lambda \eta. 1 / \text{card } (\text{prf-domain } \eta))$

shows *negligible* $(\lambda \eta. \text{ind-cca.} \text{advantage } \eta (\mathcal{A} \ \eta))$

<proof>

end

end

theory *Cryptographic-Constructions* **imports**

Elgamal

Hashed-Elgamal

RP-RF

PRF-UHF

PRF-IND-CPA

PRF-UPF-IND-CCA

begin

end

theory *Game-Based-Crypto* **imports**

Security-Spec

Cryptographic-Constructions

begin

end

A Tutorial Introduction to CryptHOL

Andreas Lochbihler

Digital Asset (Switzerland) GmbH, Zurich, Switzerland,
mail@andreas-lochbihler.de

S. Reza Sefidgar

Institute of Information Security, Department of Computer Science, ETH
Zurich, Zurich, Switzerland,
reza.sefidgar@inf.ethz.ch

Abstract

This tutorial demonstrates how cryptographic security notions, constructions, and game-based security proofs can be formalized using the CryptHOL framework. As a running example, we formalize a variant of the hash-based ElGamal encryption scheme and its IND-CPA security in the random oracle model. This tutorial assumes basic familiarity with Isabelle/HOL and standard cryptographic terminology.

3 Introduction

CryptHOL [2, 11] is a framework for constructing rigorous game-based proofs using the proof assistant Isabelle/HOL [15]. Games are expressed as probabilistic functional programs that are shallowly embedded in higher-order logic (HOL) using CryptHOL’s combinators. The security statements, both concrete and asymptotic, are expressed as Isabelle/HOL theorem statements, and their proofs are written declaratively in Isabelle’s proof language Isar [21]. This way, Isabelle mechanically checks that all definitions and statements are type-correct and each proof step is a valid logical inference in HOL. This ensures that the resulting theorems are valid in higher-order logic.

This tutorial explains the CryptHOL essentials using a simple security proof. Our running example is a variant of the hashed ElGamal encryption scheme [7]. We formalize the scheme, the indistinguishability under chosen plaintext (IND-CPA) security property, the computational Diffie-Hellman (CDH) hardness assumption [5], and the security proof in the random oracle model. This illustrates how the following aspects of a cryptographic security proof are formalized using CryptHOL:

- Game-based security definitions (CDH in §4.1 and IND-CPA in §4.4)
- Oracles (a random oracle in §4.2)
- Cryptographic schemes, both generic (the concept of an encryption scheme) and a particular instance (the hashed Elgamal scheme in §4.5)
- Security statements (concrete and asymptotic, §5.2 and §6.2)

- Reductions (from IND-CPA to CDH for hashed Elgamal in §5.1)
- Different kinds of proof steps (§5.3–5.8):
 - Using intermediate games
 - Defining failure events and applying indistinguishability-up-to lemmas
 - Equivalence transformations on games

This tutorial assumes that the reader knows the basics of Isabelle/HOL and game-based cryptography and wants to get hands-on experience with CryptHOL. The semantics behind CryptHOL’s embedding in higher-order logic and its soundness are not discussed; we refer the reader to the scientific articles for that [2, 11]. Shoup’s tutorial [19] provides a good introduction to game-based proofs. The following Isabelle features are frequently used in CryptHOL formalizations; the tutorials are available from the Documentation panel in Isabelle/jEdit.

- Function definitions (tutorials `prog-prove` and `functions`, [10]) for games and reductions
- Locales (tutorial `locales`, [1]) to modularize the formalization
- The Transfer package [9] for automating parametricity and representation independence proofs

This document is generated from a corresponding Isabelle theory file available online [13].¹ It contains this text and all examples, including the security definitions and proofs. We encourage all readers to download the latest version of the tutorial and follow the proofs and examples interactively in Isabelle/HOL. In particular, a Ctrl-click on a formal entity (function, constant, theorem name, ...) jumps to the definition of the entity.

We split the tutorial into a series of recipes for common formalization tasks. In each section, we cover a familiar cryptography concept and show how it is formalized in CryptHOL. Simultaneously, we explain the Isabelle/HOL and functional programming topics that are essential for formalizing game-based proofs.

3.1 Getting started

CryptHOL is available as part of the Archive of Formal Proofs [12]. Cryptography formalizations based on CryptHOL are arranged in Isabelle theory files that import the relevant libraries.

¹The tutorial has been added to the Archive of Formal Proofs after the release of Isabelle2018. Until the subsequent Isabelle release, the tutorial is only available in the development version at https://devel.isa-afp.org/entries/Game_Based_Crypto.html. The version for Isabelle2018 is available at http://www.andreas-lochbihler.de/pub/crypthol_tutorial.zip.

3.2 Getting started

CryptHOL is available as part of the Archive of Formal Proofs [12]. Cryptography formalizations based on CryptHOL are arranged in Isabelle theory files that import the relevant libraries.

```
theory CryptHOL-Tutorial imports  
  CryptHOL.CryptHOL  
begin
```

The file *CryptHOL.CryptHOL* is the canonical entry point into CryptHOL. For the hashed Elgamal example in this tutorial, the CryptHOL library contains everything that is needed. Additional Isabelle libraries can be imported if necessary.

4 Modelling cryptography using CryptHOL

This section demonstrates how the following cryptographic concepts are modelled in CryptHOL.

- A security property without oracles (§4.1)
- An oracle (§4.2)
- A cryptographic concept (§4.3)
- A security property with an oracle (§4.4)
- A concrete cryptographic scheme (§4.5)

4.1 Security notions without oracles: the CDH assumption

In game-based cryptography, a security property is specified using a game between a benign challenger and an adversary. The probability of an adversary to win the game against the challenger is called its advantage. A cryptographic construction satisfies a security property if the advantage for any “feasible” adversary is “negligible”. A typical security proof reduces the security of a construction to the assumed security of its building blocks. In a concrete security proof, where the security parameter is implicit, it is therefore not necessary to formally define “feasibility” and “negligibility”, as the security statement establishes a concrete relation between the advantages of specific adversaries.² We return to asymptotic security statements in §6.

A formalization of a security property must therefore specify all of the following:

²The cryptographic literature sometimes abstracts over the adversary and defines the advantage to be the advantage of the best “feasible” adversary against a game. Such abstraction would require a formalization of feasibility, for which CryptHOL currently does not offer any support. We therefore always consider the advantage of a specific adversary.

- The operations of the scheme (e.g., an algebraic group, an encryption scheme)
- The type of adversary
- The game with the challenger
- The advantage of the adversary as a function of the winning probability

For hashed Elgamal, the cyclic group must satisfy the computational Diffie-Hellman assumption. To keep the proof simple, we formalize the equivalent list version of CDH.

Definition (The list computational Diffie-Hellman game). Let \mathcal{G} be a group of order q with generator \mathbf{g} . The List Computational Diffie-Hellman (LCDH) assumption holds for \mathcal{G} if any “feasible” adversary has “negligible” probability in winning the following **LCDH game** against a challenger:

1. The challenger picks x and y randomly (and independently) from $\{0, \dots, q - 1\}$.
2. It passes \mathbf{g}^x and \mathbf{g}^y to the adversary. The adversary generates a set L of guesses about the value of \mathbf{g}^{xy} .
3. The adversary wins the game if $\mathbf{g}^{xy} \in L$.

The scheme for LCDH uses only a cyclic group. To make the LCDH formalisation reusable, we formalize the LCDH game for an arbitrary cyclic group \mathcal{G} using Isabelle’s module system based on locales. The locale *list-cdh* fixes \mathcal{G} to be a finite cyclic group that has elements of type *'grp* and comes with a generator $\mathbf{g}_{\mathcal{G}}$. Basic facts about finite groups are formalized in the CryptHOL theory *CryptHOL.Cyclic-Group*.³

```
locale list-cdh = cyclic-group  $\mathcal{G}$ 
for  $\mathcal{G} :: 'grp$  cyclic-group (structure)
begin
```

The LCDH game does not need oracles. The adversary is therefore just a probabilistic function from two group elements to a set of guesses, which are again group elements. In CryptHOL, the probabilistic nature is expressed by the adversary returning a discrete subprobability distribution over sets of guesses, as expressed by the type constructor *spmf*. (Subprobability distributions are like probability distributions except that the whole probability mass may be less than 1, i.e., some

³The syntax directive **structure** tells Isabelle that all group operations in the context of the locale refer to the group \mathcal{G} unless stated otherwise. For example, $\mathbf{g}_{\mathcal{G}}$ can be written as \mathbf{g} inside the locale.

Isabelle automatically adds the locale parameters and the assumptions on them to all definitions and lemmas inside that locale. Of course, we could have made the group \mathcal{G} an explicit argument of all functions ourselves, but then we would not benefit from Isabelle’s module system, in particular locale instantiation.

probability may be “lost”. A subprobability distribution is called lossless, written *lossless-spmf*, if its probability mass is 1.) We define the following abbreviation as a shorthand for the type of LCDH adversaries.⁴

type-synonym *'grp' adversary* = *'grp' ⇒ 'grp' ⇒ 'grp' set spmf*

The LCDH game itself is expressed as a function from the adversary \mathcal{A} to the subprobability distribution of the adversary winning. CryptHOL provides operators to express these distributions as probabilistic programs and reason about them using program logics:

- The *do* notation desugars to monadic sequencing in the monad of subprobabilities [20]. Intuitively, every line $x \leftarrow p$; samples an element x from the distribution p . The sampling is independent, unless the distribution p depends on previously sampled variables. At the end of the block, the *return-spmf* $_$ returns whether the adversary has won the game.
- *sample-uniform* n denotes the uniform distribution over the set $\{0, \dots, n - 1\}$.
- *order* \mathcal{G} denotes the order of \mathcal{G} and $([\wedge]) :: 'grp \Rightarrow nat \Rightarrow 'grp$ is the group exponentiation operator.

The LCDH game formalizes the challenger’s behavior against an adversary \mathcal{A} . In the following definition, the challenger randomly (and independently) picks two natural numbers x and y that are between 0 and \mathcal{G} ’s order and passes them to the adversary. The adversary then returns a set zs of guesses for $g^{x * y}$, where g is the generator of \mathcal{G} . The game finally returns a *boolean* that indicates whether the adversary produced a right guess. Formally, *game* \mathcal{A} is a *boolean* random variable.

definition *game* :: *'grp adversary ⇒ bool spmf* **where**

```

game  $\mathcal{A}$  = do {
   $x \leftarrow$  sample-uniform (order  $\mathcal{G}$ );
   $y \leftarrow$  sample-uniform (order  $\mathcal{G}$ );
   $zs \leftarrow$   $\mathcal{A}$  ( $\mathbf{g} [\wedge] x$ ) ( $\mathbf{g} [\wedge] y$ );
  return-spmf ( $\mathbf{g} [\wedge] (x * y) \in zs$ )
}
```

The advantage of the adversary is equivalent to its probability of winning the LCDH game. The function *spmf* :: *'a spmf ⇒ 'a ⇒ real* returns the probability of an elementary event under a given subprobability distribution.

definition *advantage* :: *'grp adversary ⇒ real*
where *advantage* \mathcal{A} = *spmf* (*game* \mathcal{A}) *True*

⁴Actually, the type of group elements has already been fixed in the locale *list-cdh* to the type variable *'grp*. Unfortunately, such fixed type variables cannot be used in type declarations inside a locale in Isabelle2018. The **type-synonym** *adversary* is therefore parametrized by a different type variable *'grp'*, but it will be used below only with *'grp*.

end

This completes the formalisation of the LCDH game and we close the locale *list-cdh* with **end**. The above definitions are now accessible under the names *game* and *advantage*. Furthermore, when we later instantiate the locale *list-cdh*, they will be specialized to the given parameters. We will return to this topic in §4.5.

4.2 A Random Oracle

A cryptographic oracle grants an adversary black-box access to a certain information or functionality. In this section, we formalize a random oracle, i.e., an oracle that models a random function with a finite codomain. In the Elgamal security proof, the random oracle represents the hash function: the adversary can query the oracle for a value and the oracle responds with the corresponding “hash”.

Like for the LCDH formalization, we wrap the random oracle in the locale *random-oracle* for modularity. The random oracle will return a *bitstring*, i.e. a list of booleans, of length *len*.

type-synonym *bitstring* = *bool list*

locale *random-oracle* =
 fixes *len* :: *nat*
begin

In CryptHOL, oracles are modeled as probabilistic transition systems that given an initial state and an input, return a subprobability distribution over the output and the successor state. The type synonym $('s, 'a, 'b) \textit{oracle}$ abbreviates $'s \Rightarrow 'a \Rightarrow ('b \times 's) \textit{spmf}$.

A random oracle accepts queries of type *'a* and generates a random bitstring of length *len*. The state of the random oracle remembers its previous responses in a mapping of type $'a \rightarrow \textit{bitstring}$. Upon a query *x*, the oracle first checks whether this query was received before. If so, the oracle returns the same answer again. Otherwise, the oracle randomly samples a bitstring of length *len*, stores it in its state, and returns it alongside with the new state.

type-synonym $'a \textit{state} = 'a \rightarrow \textit{bitstring}$

definition *oracle* :: $'a \textit{state} \Rightarrow 'a \Rightarrow (\textit{bitstring} \times 'a \textit{state}) \textit{spmf}$
where

oracle σ *x* = (case σ *x* of
 None \Rightarrow do {
 bs \leftarrow *spmf-of-set* (*nlists UNIV len*);
 return-*spmf* (*bs*, $\sigma(x \mapsto \textit{bs})$) }
 | Some *bs* \Rightarrow return-*spmf* (*bs*, σ))

Initially, the state of a random oracle is the empty map $\lambda x. \textit{None}$, as no queries have been asked. For readability, we introduce an abbreviation:

abbreviation (*input*) *initial* :: 'a state **where** *initial* \equiv *Map.empty*

This actually completes the formalization of the random oracle. Before we close the locale, we prove two technical lemmas:

1. The lemma *lossless-oracle* states that the distribution over answers and successor states is *lossless*, i.e., a full probability distribution. Many reasoning steps in game-based proofs are only valid for lossless distributions, so it is generally recommended to prove losslessness of all definitions if possible.
2. The lemma *fresh* describes random oracle's behavior when the query is fresh. This lemma makes it possible to automatically unfold the random oracle only when it is known that the query is fresh.

lemma *lossless-oracle* [*simp*]: *lossless-spmf* (*oracle* σ *x*)
<proof>

lemma *fresh*:
oracle σ *x* =
 (*do* { *bs* \leftarrow *spmf-of-set* (*nlists UNIV len*);
 return-spmf (*bs*, $\sigma(x \mapsto bs)$) })
if σ *x* = *None*
<proof>

end

Remark: Independence is the default. Note that - *spmf* represents a discrete probability distribution rather than a random variable. The difference is that every *spmf* is independent of all other *spmf*s. There is no implicit space of elementary events via which information may be passed from one random variable to the other. If such information passing is necessary, this must be made explicit in the program. That is why the random oracle explicitly takes a state of previous responses and returns the updated states. Later, whenever the random oracle is used, the user must pass the state around as needed. This also applies to adversaries that may want to store some information.

4.3 Cryptographic concepts: public-key encryption

A cryptographic concept consists of a set of operations and their functional behaviour. We have already seen two simple examples: the cyclic group in §4.1 and the random oracle in §4.2. We have formalized both of them as locales; we have not modelled their functional behavior as this is not needed for the proof. In this section, we now present a more realistic example: public-key encryption with oracle access.

A public-key encryption scheme consists of three algorithms: key generation, encryption, and decryption. They are all probabilistic and, in the most general case, they may access an oracle jointly with the adversary, e.g., a random oracle modelling a hash function. As before, the operations are modelled as parameters of a locale, *ind-cpa-pk*.

- The key generation algorithm *key-gen* outputs a public-private key pair.
- The encryption operation *encrypt* takes a public key and a plaintext of type *'plain* and outputs a ciphertext of type *'cipher*.
- The decryption operation *decrypt* takes a private key and a ciphertext and outputs a plaintext.
- Additionally, the predicate *valid-plains* tests whether the adversary has chosen a valid pair of plaintexts. This operation is needed only in the IND-CPA game definition in the next section, but we include it already here for convenience.

```

locale ind-cpa-pk =
  fixes key-gen :: ('pubkey × 'privkey, 'query, 'response) gpv
  and encrypt :: 'pubkey ⇒ 'plain ⇒ ('cipher, 'query, 'response) gpv
  and decrypt :: 'privkey ⇒ 'cipher ⇒ ('plain, 'query, 'response) gpv
  and valid-plains :: 'plain ⇒ 'plain ⇒ bool
begin

```

The three actual operations are generative probabilistic values (GPV) of type $(-, 'query, 'response) gpv$. A GPV is a probabilistic algorithm that has not yet been connected to its oracles; see the theoretical paper [2] for details. The interface to the oracle is abstracted in the two type parameters *'query* for queries and *'response* for responses. As before, we omit the specification of the functional behavior, namely that decrypting an encryption with a key pair returns the plaintext.

4.4 Security notions with oracles: IND-CPA security

In general, there are several security notions for the same cryptographic concept. For encryption schemes, an indistinguishability notion of security [8] is often used. We now formalize the notion indistinguishability under chosen plaintext attacks (IND-CPA) for public-key encryption schemes. Goldwasser et al. [18] showed that IND-CPA is equivalent to semantic security.

Definition (IND-CPA [19]). Let *key-gen*, *encrypt* and *decrypt* denote a public-key encryption scheme. The IND-CPA game is a two-stage game between the *adversary* and a *challenger*:

Stage 1 (find):

1. The challenger generates a public key pk using $key-gen$ and gives the public key to the adversary.
2. The adversary returns two messages m_0 and m_1 .
3. The challenger checks that the two messages are a valid pair of plain-texts. (For example, both messages must have the same length.)

Stage 2 (guess):

1. The challenger flips a coin b (either 0 or 1) and gives $encrypt\ pk\ m_b$ to the adversary.
2. The adversary returns a bit b' .

The adversary wins the game if his guess b' is the value of b . Let P_{win} denote the winning probability. His advantage is $|P_{win} - 1/2|$

Like with the encryption scheme, we will define the game such that the challenger and the adversary have access to a shared oracle, but the oracle is still unspecified. Consequently, the corresponding CryptHOL game is a GPV, like the operations of the abstract encryption scheme. When we specialize the definitions in the next section to the hashed Elgamal scheme, the GPV will be connected to the random oracle.

The type of adversary is now more complicated: It is a pair of probabilistic functions with oracle access, one for each stage of the game. The first computes the pair of plaintext messages and the second guesses the challenge bit. The additional *'state* parameter allows the adversary to maintain state between the two stages.

type-synonym (*'pubkey', 'plain', 'cipher', 'query', 'response', 'state*) *adversary* =
 (*'pubkey'* \Rightarrow ((*'plain'* \times *'plain'*) \times *'state', 'query', 'response'*) *gpv*)
 \times (*'cipher'* \Rightarrow *'state'* \Rightarrow (*bool, 'query', 'response'*) *gpv*)

The IND-CPA game formalization below follows the above informal definition. There are three points that need some explanation. First, this game differs from the simpler LCDH game in that it works with GPVs instead of SPMFs. Therefore, probability distributions like coin flips *coin-spmf* must be lifted from SPMFs to GPVs using the coercion *lift-spmf*. Second, the assertion *assert-gpv* (*valid-plains* $m_0\ m_1$) ensures that the pair of messages is valid. Third, the construct *TRY_ELSE* catches a violated assertion. In that case, the adversary's advantage drops to 0 because the result of the game is a coin flip, as we are in the *ELSE* branch.

fun *game* :: (*'pubkey', 'plain', 'cipher', 'query', 'response', 'state*) *adversary*
 \Rightarrow (*bool, 'query', 'response'*) *gpv*

where

game ($\mathcal{A}_1, \mathcal{A}_2$) = *TRY do* {
 (pk, sk) \leftarrow *key-gen*;
 ($(m_0, m_1), \sigma$) \leftarrow $\mathcal{A}_1\ pk$;
assert-gpv (*valid-plains* $m_0\ m_1$);
 $b \leftarrow$ *lift-spmf coin-spmf*;




Figure 1: Graphic representation of the generic IND-CPA game.

```

cipher ← encrypt pk (if b then m0 else m1);
b' ← A2 cipher σ;
Done (b' = b)
} ELSE lift-spmf coin-spmf

```

Figure 1 visualizes this game as a grey box. The dashed boxes represent parameters of the game or the locale, i.e., parts that have not yet been instantiated. The actual probabilistic program is shown on the left half, which uses the dashed boxes as sub-programs. Arrows in the grey box from the left to the right pass the contents of the variables to the sub-program. Those in the other direction bind the result of the sub-program to new variables. The arrows leaving box indicate the query-response interaction with an oracle. The thick arrows emphasize that the adversary's state is passed around explicitly. The double arrow represents the return value of the game. We will use this to define the adversary's advantage.

As the oracle is not specified in the game, the advantage, too, is parametrized by the oracle, given by the transition function $oracle :: ('s, 'query, 'response) oracle'$ and the initial state $\sigma :: 's$ its initial state. The operator $run-gpv$ connects the game with the oracle, whereby the GPV becomes an SPMF.

```

fun advantage :: ('σ, 'query, 'response) oracle' × 'σ
  ⇒ ('pubkey, 'plain, 'cipher, 'query, 'response, 'state) adversary ⇒ real
where
  advantage (oracle, σ) A = |spmf (run-gpv oracle (game A) σ) True - 1/2|
end

```

4.5 Concrete cryptographic constructions: the hashed ElGamal encryption scheme

With all the above modelling definitions in place, we are now ready to explain how concrete cryptographic constructions are expressed in CryptHOL. In general, a cryptographic construction builds a cryptographic concept from possibly several

simpler cryptographic concepts. In the running example, the hashed ElGamal cipher [7] constructs a public-key encryption scheme from a finite cyclic group and a hash function. Accordingly, the formalisation consists of three steps:

1. Import the cryptographic concepts on which the construction builds.
2. Define the concrete construction.
3. Instantiate the abstract concepts with the construction.

First, we declare a new locale that imports the two building blocks: the cyclic group from the LCDH game with namespace *lcdh* and the random oracle for the hash function with namespace *ro*. This ensures that the construction can be used for arbitrary cyclic groups. For the message space, it suffices to fix the length *len-plain* of the plaintexts.

```
locale hashed-elgamal =
  lcdh: list-cdh  $\mathcal{G}$  +
  ro: random-oracle len-plain
  for  $\mathcal{G}$  :: 'grp cyclic-group (structure)
  and len-plain :: nat
begin
```

Second, we formalize the hashed ElGamal encryption scheme. Here is the well-known informal definition.

Definition (Hashed Elgamal encryption scheme). Let G be a cyclic group of order q that has a generator g . Furthermore, let h be a hash function that maps the elements of G to bitstrings, and \oplus be the xor operator on bitstrings. The Hashed-ElGamal encryption scheme is given by the following algorithms:

Key generation Pick an element x randomly from the set $\{0, \dots, q-1\}$ and output the pair (g^x, x) , where g^x is the public key and x is the private key.

Encryption Given the public key pk and the message m , pick y randomly from the set $\{0, \dots, q-1\}$ and output the pair $(g^y, h(pk^y) \oplus m)$. Here \oplus denotes the bitwise exclusive-or of two bitstrings.

Decryption Given the private key sk and the ciphertext (α, β) , output $h(\alpha^{sk}) \oplus \beta$.

As we can see, the public key is a group element, the private key a natural number, a plaintext a bitstring, and a ciphertext a pair of a group element and a bitstring.⁵ For readability, we introduce meaningful abbreviations for these concepts.

```
type-synonym 'grp' pub-key = 'grp'
```

⁵More precisely, the private key ranges between 0 and $q-1$ and the bitstrings are of length *len-plain*. However, Isabelle/HOL's type system cannot express such properties that depend on locale parameters.

type-synonym $'grp'$ *priv-key* = *nat*
type-synonym *plain* = *bitstring*
type-synonym $'grp'$ *cipher* = $'grp' \times \textit{bitstring}$

We next translate the three algorithms into CryptHOL definitions. The definitions are straightforward except for the hashing. Since we analyze the security in the random oracle model, an application of the hash function H is modelled as a query to the random oracle using the GPV *hash*. Here, *Pause x Done* calls the oracle with query x and returns the oracle's response. Furthermore, we define the plaintext validity predicate to check the length of the adversary's messages produced by the adversary.

abbreviation $\textit{hash} :: 'grp \Rightarrow (\textit{bitstring}, 'grp, \textit{bitstring}) \textit{gpv}$
where
 $\textit{hash } x \equiv \textit{Pause } x \textit{ Done}$

definition $\textit{key-gen} :: ('grp \textit{pub-key} \times 'grp \textit{priv-key}) \textit{spmf}$
where
 $\textit{key-gen} = \textit{do} \{$
 $x \leftarrow \textit{sample-uniform } (\textit{order } \mathcal{G});$
 $\textit{return-spmf } (\mathbf{g} [\wedge] x, x)$
 $\}$

definition $\textit{encrypt} :: 'grp \textit{pub-key} \Rightarrow \textit{plain} \Rightarrow ('grp \textit{cipher}, 'grp, \textit{bitstring}) \textit{gpv}$
where
 $\textit{encrypt } \alpha \textit{ msg} = \textit{do} \{$
 $y \leftarrow \textit{lift-spmf } (\textit{sample-uniform } (\textit{order } \mathcal{G}));$
 $h \leftarrow \textit{hash } (\alpha [\wedge] y);$
 $\textit{Done } (\mathbf{g} [\wedge] y, h [\oplus] \textit{msg})$
 $\}$

definition $\textit{decrypt} :: 'grp \textit{priv-key} \Rightarrow 'grp \textit{cipher} \Rightarrow (\textit{plain}, 'grp, \textit{bitstring}) \textit{gpv}$
where
 $\textit{decrypt } x = (\lambda (\beta, \zeta). \textit{do} \{$
 $h \leftarrow \textit{hash } (\beta [\wedge] x);$
 $\textit{Done } (\zeta [\oplus] h)$
 $\})$

definition $\textit{valid-plains} :: \textit{plain} \Rightarrow \textit{plain} \Rightarrow \textit{bool}$
where
 $\textit{valid-plains } \textit{msg1 } \textit{msg2} \longleftrightarrow \textit{length } \textit{msg1} = \textit{len-plain} \wedge \textit{length } \textit{msg2} = \textit{len-plain}$

The third and last step instantiates the interface of the encryption scheme with the hashed Elgamal scheme. This specializes all definition and theorems in the locale *ind-cpa-pk* to our scheme.

sublocale $\textit{ind-cpa} : \textit{ind-cpa-pk} (\textit{lift-spmf } \textit{key-gen}) \textit{encrypt } \textit{decrypt } \textit{valid-plains} \langle \textit{proof} \rangle$

Figure 2 illustrates the instantiation. In comparison to Fig. 1, the boxes for the key generation and the encryption algorithm have been instantiated with the hashed El-




Figure 2: The IND-CPA game instantiated with the Hashed-ElGamal encryption scheme and accessing a random oracle.

gama definitions from this section. We nevertheless draw the boxes to indicate that the definitions of these algorithms has not yet been inlined in the game definition. The thick grey border around the key generation algorithm denotes the *lift-spmf* operator, which embeds the probabilistic *key-gen* without oracle access into the type of GPVs with oracle access. The oracle has also been instantiated with the random oracle *oracle* imported from *hashed-elgama*'s parent locale *random-oracle* with prefix *ro*.

5 Cryptographic proofs in CryptHOL

This section explains how cryptographic proofs are expressed in CryptHOL. We will continue our running example by stating and proving the IND-CPA security of the hashed Elgama encryption scheme under the computational Diffie-Hellman assumption in the random oracle model, using the definitions from the previous section. More precisely, we will formalize a reduction argument (§5.1) and bound the IND-CPA advantage using the CDH advantage. We will *not* formally state the result that CDH hardness in the cyclic group implies IND-CPA security, which quantifies over all feasible adversaries—to that end, we would have to formally define feasibility, for which CryptHOL currently does not offer any support.

The actual proof of the bound consists of several game transformations. We will focus on those steps that illustrate common steps in cryptographic proofs (§5.3–§5.8)

5.1 The reduction

The security proof involves a reduction argument: We will derive a bound on the advantage of an arbitrary adversary in the IND-CPA game *game* for hashed Elgama that depends on another adversary's advantage in the LCDH game *game* of the




Figure 3: The reduction for the Elgamal security proof.

underlying group. The reduction transforms every IND-CPA adversary \mathcal{A} into a LCDH adversary *elgamal-reduction* \mathcal{A} , using \mathcal{A} as a black box. In more detail, it simulates an execution of the IND-CPA game including the random oracle. At the end of the game, the reduction outputs the set of queries that the adversary has sent to the random oracle. The reduction works as follows given a two part IND-CPA adversary $\mathcal{A} = (\mathcal{A}_1, \mathcal{A}_2)$ (Figure 3 visualizes the reduction as the dotted box):

1. It receives two group elements α and β from the LCDH challenger.
2. The reduction passes α to the adversary as the public key and runs \mathcal{A}_1 to get messages m_1 and m_2 . The adversary is given access to the random oracle with the initial state λx . *None*.
3. The assertion checks that the adversary returns two valid plaintexts, i.e., m_1 and m_2 are strings of length *len-plain*.
4. Instead of actually performing an encryption, the reduction generates a random bitstring h of length *len-plain* (*nlists UNIV len-plain* denotes the set of all bitstrings of length *len-plain* and *spmf-of-set* converts the set into a uniform distribution over the set.)
5. The reduction passes (β, h) as the challenge ciphertext to the adversary in the second phase of the IND-CPA game.
6. The actual guess b' of the adversary is ignored; instead the reduction returns the set *dom s'* of all queries that the adversary made to the random oracle as its guess for the CDH game.
7. If any of the steps after the first phase fails, the reduction's guess is the set *dom s* of oracle queries made during the first phase.


```

fun elgamal-reduction
  :: ('grp pub-key, plain, 'grp cipher, 'grp, bitstring, 'state) ind-cpa.adversary
  ⇒ 'grp lcdh.adversary
where
  elgamal-reduction ( $\mathcal{A}_1, \mathcal{A}_2$ )  $\alpha \beta = do$  {
    (( $m_1, m_2$ ),  $\sigma$ ),  $s$  ← exec-gpv ro.oracle ( $\mathcal{A}_1 \alpha$ ) ro.initial;
    TRY do {
      - :: unit ← assert-spmf (valid-plains  $m_1 m_2$ );
       $h$  ← spmf-of-set (nlists UNIV len-plain);
      ( $b', s'$ ) ← exec-gpv ro.oracle ( $\mathcal{A}_2 (\beta, h) \sigma$ )  $s$ ;
      return-spmf (dom  $s'$ )
    } ELSE return-spmf (dom  $s$ )
  }

```

5.2 Concrete security statement

A concrete security statement in CryptHOL has the form: Subject to some side conditions for the adversary \mathcal{A} , the advantage in one game is bounded by a function of the transformed adversary's advantage in a different game.⁶

theorem *concrete-security*:
assumes *side conditions for* \mathcal{A}
shows *advantage*₁ $\mathcal{A} \leq f$ (*advantage*₂ (*reduction* \mathcal{A}))

For the hashed Elgamal scheme, the theorem looks as follows, i.e., the function f is the identity function.

theorem *concrete-security-elgamal*:
assumes *lossless: ind-cpa.lossless* \mathcal{A}
shows *ind-cpa.advantage* (*ro.oracle*, *ro.initial*) $\mathcal{A} \leq$ *lcdh.advantage* (*elgamal-reduction* \mathcal{A})

Such a statement captures the essence of a concrete security proof. For if there was a feasible adversary \mathcal{A} with non-negligible advantage against the *game*, then *elgamal-reduction* \mathcal{A} would be an adversary against the *game* with at least the same advantage. This implies the existence of an adversary with non-negligible advantage against the cryptographic primitive that was assumed to be secure. What we cannot state formally is that the transformed adversary *elgamal-reduction* \mathcal{A} is feasible as we have not formalized the notion of feasibility. The readers of the formalization must convince themselves that the reduction preserves feasibility.

In the case of *elgamal-reduction*, this should be obvious from the definition (given the theorem's side condition) as the reduction does nothing more than sampling and redirecting data.

⁶A security proof often involves several reductions. The bound then depends on several advantages, one for each reduction.

Our proof for the concrete security theorem needs the side condition that the adversary is lossless. Losslessness for adversaries is similar to losslessness for subprobability distributions. It ensures that the adversary always terminates and returns an answer to the challenger. For the IND-CPA game, we define losslessness as follows:

definition (in *ind-cpa-pk*) *lossless*
 $:: ('pubkey, 'plain, 'cipher, 'query, 'response, 'state) adversary \Rightarrow bool$
where
 $lossless = (\lambda (\mathcal{A}_1, \mathcal{A}_2). (\forall pk. lossless-gpv \mathcal{I}\text{-full} (\mathcal{A}_1 pk))$
 $\wedge (\forall cipher \sigma. lossless-gpv \mathcal{I}\text{-full} (\mathcal{A}_2 cipher \sigma)))$

So now let's start with the proof.

proof –

As a preparatory step, we split the adversary \mathcal{A} into its two phases \mathcal{A}_1 and \mathcal{A}_2 . We could have made the two phases explicit in the theorem statement, but our form is easier to read and use. We also immediately decompose the losslessness assumption on \mathcal{A} .⁷

obtain $\mathcal{A}_1 \mathcal{A}_2$ **where** \mathcal{A} [simp]: $\mathcal{A} = (\mathcal{A}_1, \mathcal{A}_2)$ **by** (cases \mathcal{A})
from *lossless* **have** *lossless1* [simp]: $\wedge pk. lossless-gpv \mathcal{I}\text{-full} (\mathcal{A}_1 pk)$
and *lossless2* [simp]: $\wedge \sigma cipher. lossless-gpv \mathcal{I}\text{-full} (\mathcal{A}_2 \sigma cipher)$
by(auto simp add: *ind-cpa.lossless-def*)

5.3 Recording adversary queries

As can be seen in Fig. 2, both the adversary and the encryption of the challenge ciphertext use the random oracle. The reduction, however, returns only the queries that the adversary makes to the oracle (in Fig. 3, h is generated independently of the random oracle). To bridge this gap, we introduce an *interceptor* between the adversary and the oracle that records all adversary's queries.

define *interceptor* $:: 'grp set \Rightarrow 'grp \Rightarrow (bitstring \times 'grp set, -, -) gpv$
where
 $interceptor \sigma x = (do \{$
 $h \leftarrow hash x;$
 $Done (h, insert x \sigma)$
 $\})$ **for** σx

We integrate this interceptor into the *game* using the *inline* function as illustrated in Fig. 4 and name the result *game₀*.

define *game₀* **where**

⁷Later in the proof, we will often prove losslessness of the definitions in the proof. We will not show them in this document, but they are in the Isabelle sources from which this document is generated.




Figure 4: The IND-CPA game after expanding the key generation algorithm’s definition and inlining the query-recording hash oracle. The red boxes represent the inline operator.

```

game0 = TRY do {
  (pk, -) ← lift-spmf key-gen;
  ((m1, m2), σ), s ← inline interceptor (A1 pk) {};
  assert-gpv (valid-plains m1 m2);
  b ← lift-spmf coin-spmf;
  c ← encrypt pk (if b then m1 else m2);
  (b', s') ← inline interceptor (A2 c σ) s;
  Done (b' = b)
} ELSE lift-spmf coin-spmf

```

We claim that the above modifications do not affect the output of the IND-CPA game at all. This might seem obvious since we are only logging the adversary’s queries without modifying them. However, in a formal proof, this needs to be precisely justified.

More precisely, we have been very careful that the two games *game* \mathcal{A} and *game*₀ have identical structure. They differ only in that *game*₀ uses the adversary $(\lambda pk. inline\ interceptor\ (\mathcal{A}_1\ pk)\ \emptyset, \lambda cipher\ \sigma. inline\ interceptor\ (\mathcal{A}_2\ cipher\ \sigma))$ instead of \mathcal{A} . The formal justification for this replacement happens in two steps:

1. We replace the oracle transformer *interceptor* with *id-oracle*, which merely passes queries and results to the oracle.
2. Inlining the identity oracle transformer *id-oracle* does not change an adversary and can therefore be dropped.

The first step is automated using Isabelle’s Transfer package [9], which is based on Mitchell’s representation independence [14]. The replacement is controlled by so-called transfer rules of the form $R\ x\ y$ which indicates that x shall replace y ; the correspondence relation R captures the kind of replacement. The *transfer* proof method then constructs a constraint system with one constraint for each atom in the

proof goal where the correspondence relation and the replacement are unknown. It then tries to solve the constraint system using the rules that have been declared with the attribute `[transfer-rule]`. Atoms that do not have a suitable transfer rule are not changed and their correspondence relation is instantiated with the identity relation (`=`).

The second step is automated using Isabelle’s simplifier.

In the example, the crucial change happens in the state of the oracle transformer: *interceptor* records all queries in a set whereas *id-oracle* has no state, which is modelled with the singleton type *unit*. To capture the change, we define the correspondence relation *cr* on the states of the oracle transformers. (As we are in the process of adding this state, this state is irrelevant and *cr* is therefore always true. We nevertheless have to make an explicit definition such that Isabelle does not automatically beta-reduce terms, which would confuse *transfer*.) We then prove that it relates the initial states and that *cr* is a bisimulation relation for the two oracle transformers; see [2] for details. The bisimulation proof itself is automated, too: A bit of term rewriting (**unfolding**) makes the two oracle transformers structurally identical except for the state update function. Having proved that the state update function $\lambda\sigma. \sigma$ is a correct replacement for *insert* w.r.t. *cr*, the *transfer-prover* then lifts this replacement to the bisimulation rule. Here, *transfer-prover* is similar to *transfer* except that it works only for transfer rules and builds the constraint system only for the term to be replaced.

The theory source of this tutorial contains a step-by-step proof to illustrate how transfer works.

```
{ define cr :: unit ⇒ 'grp set ⇒ bool where cr σ σ' = True for σ σ'
  have [transfer-rule]: cr () {} by (simp add: cr-def) — initial states
  have [transfer-rule]: ((=) == => cr == => cr) (λσ. σ) insert — state update
    by (simp add: rel-fun-def cr-def)
  have [transfer-rule]: — cr is a bisimulation for the oracle transformers
    (cr == => (=) == => rel-gpv (rel-prod (=) cr) (=)) id-oracle interceptor
  unfolding interceptor-def [abs-def] id-oracle-def [abs-def] bind-gpv-Pause bind-rpv-Done
  by transfer-prover
  have ind-cpa.game ℳ = game0 unfolding game0-def ℳ ind-cpa.game.simps
  by transfer (simp add: bind-map-gpv o-def ind-cpa.game.simps split-def)
}
```

5.4 Equational program transformations

Before we move on, we need to simplify *game₀* and inline a few of the definitions. All these simplifications are equational program transformations, so the Isabelle simplifier can justify them. We combine the *interceptor* with the random oracle *oracle* into a new oracle *oracle'* with which the adversary interacts.

```
define oracle' :: 'grp set × ('grp → bitstring) ⇒ 'grp ⇒ -
  where oracle' = (λ(s, σ) x. do {
    (h, σ') ← case σ x of
```

```

None ⇒ do {
  bs ← spmf-of-set (nlists UNIV len-plain);
  return-spmf (bs, σ(x ↦ bs)) }
| Some bs ⇒ return-spmf (bs, σ);
return-spmf (h, insert x s, σ')
})
have *: exec-gpv ro.oracle (inline interceptor  $\mathcal{A}$  s) σ =
map-spmf (λ(a, b, c). ((a, b), c)) (exec-gpv oracle'  $\mathcal{A}$  (s, σ)) for  $\mathcal{A}$  σ s
by(simp add: interceptor-def oracle'-def ro.oracle-def Let-def
exec-gpv-inline exec-gpv-bind o-def split-def cong del: option.case-cong-weak)

```

We also want to inline the key generation and encryption algorithms, push the *TRY_ELSE_* towards the assertion (which is possible because the adversary is lossless by assumption), and rearrange the samplings a bit. The latter is automated using *monad-normalisation* [17].⁸

```

have game0: run-gpv ro.oracle game0 ro.initial = do {
  x ← sample-uniform (order  $\mathcal{G}$ );
  y ← sample-uniform (order  $\mathcal{G}$ );
  b ← coin-spmf;
  ((msg1, msg2), σ), (s, s-h) ←
  exec-gpv oracle' ( $\mathcal{A}_1$  (g [^] x)) ({}, ro.initial);
  TRY do {
    - :: unit ← assert-spmf (valid-plains msg1 msg2);
    (h, s-h') ← ro.oracle s-h (g [^] (x * y));
    let cipher = (g [^] y, h [⊕] (if b then msg1 else msg2));
    (b', (s', s-h')) ← exec-gpv oracle' ( $\mathcal{A}_2$  cipher σ) (s, s-h');
    return-spmf (b' = b)
  } ELSE do {
    b ← coin-spmf;
    return-spmf b
  }
}
including monad-normalisation
by(simp add: game0-def key-gen-def encrypt-def * exec-gpv-bind bind-map-spmf as-
sert-spmf-def
try-bind-assert-gpv try-gpv-bind-lossless split-def o-def if-distrib lcdh.nat-pow-pow)

```

This call to Isabelle’s simplifier may look complicated at first, but it can be constructed incrementally by adding a few theorems and looking at the resulting goal state and searching for suitable theorems using **find-theorems**. As always in Isabelle, some intuition and knowledge about the library of lemmas is crucial.

- We knew that the definitions *game₀-def*, *key-gen-def*, and *encrypt-def* should be unfolded, so they are added first to the simplifier’s set of rewrite rules.

⁸The tool *monad-normalisation* augments Isabelle’s simplifier with a normalization procedure for commutative monads based on higher-order ordered rewriting. It can also commute across control structures like *if* and *case*. Although it is not complete as a decision procedure (as the normal forms are not unique), it usually works in practice.




Figure 5: The IND-CPA game after flattening. The blue box around the encryption algorithm and the random oracle represents the expanded definition of them.

- The equations *exec-gpv-bind*, *try-bind-assert-gpv*, and *try-gpv-bind-lossless* ensure that the operator *exec-gpv*, which connects the *game₀* with the random oracle, is distributed over the sequencing. Together with $*$, this gives the adversary access to *oracle'* instead of the interceptor and the random oracle, and makes the call to the random oracle in the encryption of the chosen message explicit.
- The theorem *lcdh.nat-pow-pow* rewrites the iterated exponentiation $(\mathbf{g} [\wedge] x) [\wedge] y$ to $\mathbf{g} [\wedge] (x * y)$.
- The other theorems *bind-map-spmf*, *assert-spmf-def*, *split-def*, *o-def*, and *if-distrib* take care of all the boilerplate code that makes all these transformations type-correct. These theorems often have to be used together.

Note that the state of the oracle *oracle'* is changed between \mathcal{A}_1 and \mathcal{A}_2 . Namely, the random oracle's part *s-h* may change when the chosen message is encrypted, but the state that records the adversary's queries *s* is passed on unchanged.

5.5 Capturing a failure event

Suppose that two games behave the same except when a so-called failure event occurs [19]. Then the chance of an adversary distinguishing the two games is bounded by the probability of the failure event. In other words, the simulation of the reduction is allowed to break if the failure event occurs. In the running example, such an argument is a key step to derive the bound on the adversary's advantage. But to reason about failure events, we must first introduce them into the games we consider. This is because in CryptHOL, the probabilistic programs describe probability distributions over what they return (*return-spmf*). The variables that are used internally in the program are not accessible from the outside, i.e., there is

no memory to which these are written. This has the advantage that we never have to worry about the names of the variables, e.g., to avoid clashes. The drawback is that we must explicitly introduce all the events that we are interested in.

Introducing a failure event into a game is straightforward. So far, the games *game* and *game₀* simply denoted the probability distribution of whether the adversary has guessed right. For hashed Elgamal, the simulation breaks if the adversary queries the random oracle with the same query $\mathbf{g} [\wedge] (x * y)$ that is used for encrypting the chosen message m_b . So we simply change the return type of the game to return whether the adversary guessed right *and* whether the failure event has occurred. The next definition *game₁* does so. (Recall that *oracle'* stores in its first state component s the queries by the adversary.) In preparation of the next reasoning step, we also split off the first two samplings, namely of x and y , and make them parameters of *game₁*.

```
define game1 :: nat ⇒ nat ⇒ (bool × bool) spmf
where game1 x y = do {
  b ← coin-spmf;
  (((m1, m2), σ), (s, s-h)) ← exec-gpv oracle' (A1 (g [∧] x)) ({} , ro.initial);
  TRY do {
    - :: unit ← assert-spmf (valid-plains m1 m2);
    (h, s-h') ← ro.oracle s-h (g [∧] (x * y));
    let c = (g [∧] y, h [⊕] (if b then m1 else m2));
    (b', (s', s-h')) ← exec-gpv oracle' (A2 c σ) (s, s-h');
    return-spmf (b' = b, g [∧] (x * y) ∈ s')
  } ELSE do {
    b ← coin-spmf;
    return-spmf (b, g [∧] (x * y) ∈ s)
  }
} for x y
```

It is easy to prove that *game₀* combined with the random oracle is a projection of *game₁* with the sampling added, as formalized in *game₀-game₁*.

```
let ?sample = λf :: nat ⇒ nat ⇒ - spmf. do {
  x ← sample-uniform (order G);
  y ← sample-uniform (order G);
  f x y }
have game0-game1:
  run-gpv ro.oracle game0 ro.initial = map-spmf fst (?sample game1)
by(simp add: game0 game1-def o-def split-def map-try-spmf map-scale-spmf)
```

5.6 Game hop based on a failure event

A game hop based on a failure event changes one game into another such that they behave identically unless the failure event occurs. The *fundamental-lemma* bounds the absolute difference between the two games by the probability of the failure event. In the running example, we would like to avoid querying the random oracle when encrypting the chosen message. The next game *game₂* is identical except that

the call to the random oracle *oracle* is replaced with sampling a random bitstring.⁹

```

define game2 :: nat ⇒ nat ⇒ (bool × bool) spmf
where game2 x y = do {
  b ← coin-spmf;
  (((m1, m2), σ), (s, s-h)) ← exec-gpv oracle' (A1 (g [^] x)) ({} , ro.initial);
  TRY do {
    - :: unit ← assert-spmf (valid-plains m1 m2);
    h ← spmf-of-set (nlists UNIV len-plain);
    — We do not query the random oracle for g [^] (x * y), but instead sample a random
    bitstring h directly. So the rest differs from game1 only if the adversary queries g [^] (x *
    y).
    let cipher = (g [^] y, h [⊕] (if b then m1 else m2));
    (b', (s', s-h')) ← exec-gpv oracle' (A2 cipher σ) (s, s-h);
    return-spmf (b' = b, g [^] (x * y) ∈ s')
  } ELSE do {
    b ← coin-spmf;
    return-spmf (b, g [^] (x * y) ∈ s)
  }
} for x y

```

To apply the *fundamental-lemma*, we first have to prove that the two games are indeed the same except when the failure event occurs.

```

have rel-spmf (λ(win, bad) (win', bad')). bad = bad' ∧ (¬ bad' → win = win') (game2
x y) (game1 x y) for x y
proof —

```

This proof requires two invariants on the state of *oracle'*. First, $s = \text{dom } s-h$. Second, s only becomes larger. The next two statements capture the two invariants:

```

interpret inv-oracle': callee-invariant-on oracle' (λ(s, s-h). s = dom s-h)  $\mathcal{I}$ -full
by unfold-locales(auto simp add: oracle'-def split: option.split-asm if-split)
interpret bad: callee-invariant-on oracle' (λ(s, -). z ∈ s)  $\mathcal{I}$ -full for z
by unfold-locales(auto simp add: oracle'-def)

```

First, we identify a bisimulation relation $?X$ between the different states of *oracle'* for the second phase of the game. Namely, the invariant $s = \text{dom } s-h$ holds, the set of queries are the same, and the random oracle's state (a map from queries to responses) differs only at the point $\mathbf{g} [^] (x * y)$.

```

let ?X = λ(s, s-h) (s', s-h'). s = dom s-h ∧ s' = s ∧ s-h = s-h'(g [^] (x * y) := None)

```

Then, we can prove that $?X$ really is a bisimulation for *oracle'* except when the failure event occurs. The next statement expresses this.

```

let ?bad = λ(s, s-h). g [^] (x * y) ∈ s
let ?R = (λ(a, s1') (b, s2')). ?bad s1' = ?bad s2' ∧ (¬ ?bad s2' → a = b ∧ ?X s1' s2')
have bisim: rel-spmf ?R (oracle' s1 plain) (oracle' s2 plain)

```

⁹In Shoup's terminology [19], such a step makes (a gnome sitting inside) the random oracle forgetting the query.

if $?X\ s1\ s2$ **for** $s1\ s2$ **plain using that**
by(*auto split: prod.splits intro!: rel-spmf-bind-refl simp add: oracle'-def rel-spmf-return-spmf2*
fun-upd-twist split: option.split dest!: fun-upd-eqD)
have *inv: callee-invariant oracle' ?bad*
— Once the failure event has happened, it will not be forgotten any more.
by(*unfold-locales*)(*auto simp add: oracle'-def split: option.split-asm*)

Now we are ready to prove that the two games $game_1$ and $game_2$ are sufficiently similar. The Isar proof now switches into an **apply** script that manipulates the goal state directly. This is sometimes convenient when it would be too cumbersome to spell out every intermediate goal state.

show *?thesis*
unfolding *game₁-def game₂-def*
— Peel off the first phase of the game using the structural decomposition rules *rel-spmf-bind-refl* and *rel-spmf-try-spmf*.
apply(*clarsimp intro!: rel-spmf-bind-refl simp del: bind-spmf-const*)
apply(*rule rel-spmf-try-spmf*)
subgoal *TRY for b m₁ m₂ σ s s-h*
apply(*rule rel-spmf-bind-refl*)
— Exploit that in the first phase of the game, the set s of queried strings and the map of the random oracle $s-h$ are updated in lock step, i.e., $s = \text{dom } s-h$.
apply(*drule inv-oracle'.exec-gpv-invariant; clarsimp*)
— Has the adversary queried the random oracle with $\mathbf{g} \text{ } [\wedge] (x * y)$ during the first phase?
apply(*cases $\mathbf{g} \text{ } [\wedge] (x * y) \in s$*)
subgoal *True* — Then the failure event has already happened and there is nothing more to do. We just have to prove that the two games on both sides terminate with the same probability.
by(*auto intro!: rel-spmf-bindI1 rel-spmf-bindI2 lossless-exec-gpv[where $\mathcal{S} = \mathcal{S}\text{-full}$] dest!: bad.exec-gpv-invariant*)
subgoal *False* — Then let's see whether the adversary queries $\mathbf{g} \text{ } [\wedge] (x * y)$ in the second phase. Thanks to *ro.fresh*, the call to the random oracle simplifies to sampling a random bitstring.
apply(*clarsimp iff del: domIff simp add: domIff ro.fresh intro!: rel-spmf-bind-refl*)
apply(*rule rel-spmf-bindI[where $R = ?R$]*)
— The lemma *exec-gpv-oracle-bisim-bad-full* lifts the bisimulation for *oracle'* to the adversary \mathcal{A}_2 interacting with *oracle'*.
apply(*rule exec-gpv-oracle-bisim-bad-full[OF - - bisim inv inv]*)
apply(*auto simp add: fun-upd-idem*)
done
done
subgoal *ELSE by(rule rel-spmf-refl) clarsimp*
done
qed

Now we can add the sampling of x and y in front of $game_1$ and $game_2$, apply the *fundamental-lemma*.

hence *rel-spmf* ($\lambda (win, bad) (win', bad'). (bad \longleftrightarrow bad') \wedge (\neg bad' \longrightarrow win \longleftrightarrow win')$)
(*?sample game₂*) (*?sample game₁*)
by(*intro rel-spmf-bind-refl*)

hence $|measure (measure\text{-}spmf (?sample\ game_2)) \{(win, -). win\} - measure (measure\text{-}spmf (?sample\ game_1)) \{(win, -). win\}|$
 $\leq measure (measure\text{-}spmf (?sample\ game_2)) \{(-, bad). bad\}$
unfolding *split-def* **by** (*rule fundamental-lemma*)
moreover

The *fundamental-lemma* is written in full generality for arbitrary events, i.e., sets of elementary events. But in this formalization, the events of interest (correct guess and failure) are elementary events. We therefore transform the above statement to measure the probability of elementary events using *spmf*.

have $measure (measure\text{-}spmf (?sample\ game_2)) \{(win, -). win\} = spmf (map\text{-}spmf\ fst (?sample\ game_2))\ True$
and $measure (measure\text{-}spmf (?sample\ game_1)) \{(win, -). win\} = spmf (map\text{-}spmf\ fst (?sample\ game_1))\ True$
and $measure (measure\text{-}spmf (?sample\ game_2)) \{(-, bad). bad\} = spmf (map\text{-}spmf\ snd (?sample\ game_2))\ True$
unfolding *spmf-conv-measure-spmf measure-map-spmf* **by** (*auto simp add: vimage-def split-def*)
ultimately have *hop12*:
 $|spmf (map\text{-}spmf\ fst (?sample\ game_2))\ True - spmf (map\text{-}spmf\ fst (?sample\ game_1))\ True|$
 $\leq spmf (map\text{-}spmf\ snd (?sample\ game_2))\ True$
by *simp*

5.7 Optimistic sampling: the one-time-pad

This step is based on the one-time-pad, which is an instance of optimistic sampling. If two runs of the two games in an optimistic sampling step would use the same random bits, then their results would be different. However, if the adversary's choices are independent of the random bits, we may relate runs that use different random bits, as in the end, only the probabilities have to match. The previous game hop from *game*₁ to *game*₂ made the oracle's responses in the second phase independent from the encrypted ciphertext. So we can now change the bits used for encrypting the chosen message and thereby make the ciphertext independent of the message.

To that end, we parametrize *game*₂ by the part that does the optimistic sampling and call this parametrized version *game*₃.

define $game_3 :: (bool \Rightarrow bitstring \Rightarrow bitstring \Rightarrow bitstring\ spmf) \Rightarrow nat \Rightarrow nat \Rightarrow (bool \times bool)\ spmf$
where $game_3\ f\ x\ y = do \{$
 $b \leftarrow coin\text{-}spmf;$
 $((m_1, m_2), \sigma), (s, s\text{-}h) \leftarrow exec\text{-}gpv\ oracle' (\mathcal{A}_1 (\mathbf{g} [\wedge] x)) (\{\}, ro.initial);$
 $TRY\ do \{$
 $- :: unit \leftarrow assert\text{-}spmf (valid\text{-}plains\ m_1\ m_2);$
 $h' \leftarrow f\ b\ m_1\ m_2;$
 $let\ cipher = (\mathbf{g} [\wedge] y, h');$
 $(b', (s', s\text{-}h')) \leftarrow exec\text{-}gpv\ oracle' (\mathcal{A}_2\ cipher\ \sigma) (s, s\text{-}h);$
 $\}$
 $\}$

```

    return-spmf (b' = b, g [^] (x * y) ∈ s')
  } ELSE do {
    b ← coin-spmf;
    return-spmf (b, g [^] (x * y) ∈ s)
  }
} for f x y

```

Clearly, if we plug in the appropriate function $?f$, then we get $game_2$:

```

let ?f = λ b m1 m2. map-spmf (λ h. (if b then m1 else m2) [⊕] h) (spmf-of-set (nlists UNIV len-plain))
have game2-game3: game2 x y = game3 ?f x y for x y
by(simp add: game2-def game3-def Let-def bind-map-spmf xor-list-commute o-def)

```

CryptHOL's *one-time-pad* lemma now allows us to remove the exclusive or with the chosen message, because the resulting distributions are the same. The proof is slightly non-trivial because the one-time-pad lemma holds only if the xor'ed bitstrings have the right length, which the assertion *valid-plains* ensures. The congruence rules *try-spmf-cong* *bind-spmf-cong* [*OF refl*] *if-cong* [*OF refl*] extract this information from the program of the game.

```

let ?f' = λ b m1 m2. spmf-of-set (nlists UNIV len-plain)
have game3: game3 ?f x y = game3 ?f' x y for x y
by(auto intro!: try-spmf-cong bind-spmf-cong[OF refl] if-cong[OF refl]
    simp add: game3-def split-def one-time-pad valid-plains-def
    simp del: map-spmf-of-set-inj-on bind-spmf-const split: if-split)

```

The rest of the proof consists of simplifying $game_3 ?f'$. The steps are similar to what we have shown before, so we do not explain them in detail. The interested reader can look at them in the theory file from which this document was generated. At a high level, we see that there is no need to track the adversary's queries in $game_2$ or $game_3$ any more because this information is already stored in the random oracle's state. So we change the *oracle'* back into *oracle* using the Transfer package. With a bit of rewriting, the result is then the *game* for the adversary *elgamal-reduction* \mathcal{A} . Moreover, the guess b' of the adversary is independent of b in $game_3 ?f$, so the first boolean returned by $game_3 ?f'$ is just a coin flip.

```

have game3-bad: map-spmf snd (?sample (game3 ?f')) = lcdh.game (elgamal-reduction  $\mathcal{A}$ )
have game3-guess: map-spmf fst (game3 ?f' x y) = coin-spmf for x y

```

5.8 Combining several game hops

Finally, we combine all the (in)equalities of the previous steps to obtain the desired bound using the lemmas for reasoning about reals from Isabelle's library.

```

have ind-cpa.advantage (ro.oracle, ro.initial)  $\mathcal{A}$  = |spmf (map-spmf fst (?sample game1)) True - 1 / 2|
using ind-cpa-game-eq-game0 by(simp add: game0-game1 o-def)

```

```

also have ... = |1 / 2 - spmf (map-spmf fst (?sample game1)) True|
  by(simp add: abs-minus-commute)
also have 1 / 2 = spmf (map-spmf fst (?sample game2)) True
  by(simp add: game2-game3 game3 o-def game3-guess spmf-of-set)
also have |... - spmf (map-spmf fst (?sample game1)) True| ≤ spmf (map-spmf snd
  (?sample game2)) True
  by(rule hop12)
also have ... = lcdh.advantage (elgamal-reduction  $\mathcal{A}$ )
  by(simp add: game2-game3 game3 game3-bad lcdh.advantage-def o-def del: map-bind-spmf)
finally show ?thesis .

```

This completes the concrete proof and we can end the locale *hashed-elgamal*.

qed

end

6 Asymptotic security

An asymptotic security statement can be easily derived from a concrete security theorem. This is done in two steps: First, we have to introduce a security parameter η into the definitions and assumptions. Only then can we state asymptotic security. The proof is easy given the concrete security theorem.

6.1 Introducing a security parameter

Since all our definitions were done in locales, it is easy to introduce a security parameter after the fact. To that end, we define copies of all locales where their parameters now take the security parameter as an additional argument. We illustrate it for the locale *ind-cpa-pk*.

The **sublocale** command brings all the definitions and theorems of the original *ind-cpa-pk* into the copy and adds the security parameter where necessary. The type *security* is a synonym for *nat*.

```

locale ind-cpa-pk' =
  fixes key-gen :: security ⇒ ('pubkey × 'privkey, 'query, 'response) gpv
  and encrypt :: security ⇒ 'pubkey ⇒ 'plain ⇒ ('cipher, 'query, 'response) gpv
  and decrypt :: security ⇒ 'privkey ⇒ 'cipher ⇒ ('plain, 'query, 'response) gpv
  and valid-plains :: security ⇒ 'plain ⇒ 'plain ⇒ bool
begin
sublocale ind-cpa-pk key-gen  $\eta$  encrypt  $\eta$  decrypt  $\eta$  valid-plains  $\eta$  for  $\eta$  (proof)
end

```

We do so similarly for *list-cdh*, *random-oracle*, and *hashed-elgamal*.

```

locale hashed-elgamal' =
  lcdh: list-cdh'  $\mathcal{G}$  +

```

```

ro: random-oracle' len-plain
for  $\mathcal{G} :: security \Rightarrow 'grp\ cyclic-group$ 
and len-plain :: security  $\Rightarrow nat$ 
begin
sublocale hashed-elgamal  $\mathcal{G} \ \eta\ len-plain\ \eta$  for  $\eta$   $\langle proof \rangle$ 

```

6.2 Asymptotic security statements

For asymptotic security statements, CryptHOL defines the predicate *negligible*. It states that the given real-valued function approaches 0 faster than the inverse of any polynomial. A concrete security statement translates into an asymptotic one as follows:

- All advantages in the bound become negligibility assumptions.
- All side conditions of the concrete security theorems remain assumptions, but wrapped into an *eventually* statement. This expresses that the side condition holds eventually, i.e., there is a security parameter from which on it holds.
- The conclusion is that the bounded advantage is *negligible*.

theorem *asymptotic-security-elgamal*:

```

assumes negligible ( $\lambda \eta. lcdh.advantage\ \eta\ (elgamal-reduction\ \eta\ (\mathcal{A}\ \eta))$ )
and eventually ( $\lambda \eta. ind-cpa.lossless\ (\mathcal{A}\ \eta)$ ) at-top
shows negligible ( $\lambda \eta. ind-cpa.advantage\ \eta\ (ro.oracle\ \eta, ro.initial)\ (\mathcal{A}\ \eta)$ )

```

The proof is canonical, too: Using the lemmas about *negligible* and Eberl's library for asymptotic reasoning [6], we transform the asymptotic statement into a concrete one and then simply use the concrete security statement.

$\langle proof \rangle$

end

References

- [1] C. Ballarín. Locales: A module system for mathematical theories. *Journal of Automated Reasoning*, 52(2):123–153, Feb 2014.
- [2] D. A. Basin, A. Lochbihler, and S. R. Sefidgar. Crypthol: Game-based proofs in higher-order logic. *IACR Cryptology ePrint Archive*, 2017:753, 2017.
- [3] M. Bellare, A. Boldyreva, and S. Micali. Public-key encryption in a multi-user setting: Security proofs and improvements. In B. Preneel, editor, *Advances in Cryptology (EUROCRYPT 2000)*, volume 1807 of *Lecture Notes in Computer Science*, pages 259–274. Springer Berlin Heidelberg, 2000.

- [4] M. Bellare and P. Rogaway. The security of triple encryption and a framework for code-based game-playing proofs. In *EUROCRYPT 2006*, volume 4004 of *LNCS*, pages 409–426. Springer, 2006.
- [5] W. Diffie and M. Hellman. New directions in cryptography. *IEEE Transactions on Information Theory*, 22(6):644–654, 1976.
- [6] M. Eberl. Landau symbols. *Archive of Formal Proofs*, Jul 2015. http://isa-afp.org/entries/Landau_Symbols.html, Formal proof development.
- [7] T. ElGamal. A public key cryptosystem and a signature scheme based on discrete logarithms. *IEEE Transactions on Information Theory*, 31(4):469–472, 1985.
- [8] S. Goldwasser and S. Micali. Probabilistic encryption & how to play mental poker keeping secret all partial information. In *ACM Symposium on Theory of Computing (STOC 1982), Proceedings*, pages 365–377, 1982.
- [9] B. Huffman and O. Kunčar. Lifting and Transfer: A modular design for quotients in Isabelle/HOL. In *Certified Programs and Proofs (CPP 2013), Proceedings*, pages 131–146, 2013.
- [10] A. Krauss. *Automating Recursive Definitions and Termination Proofs in Higher-Order Logic*. Dissertation, Technische Universität München, 2009.
- [11] A. Lochbihler. Probabilistic functions and cryptographic oracles in higher order logic. In *European Symposium on Programming (ESOP 2016), Proceedings*, pages 503–531, 2016.
- [12] A. Lochbihler. CryptHOL. *Archive of Formal Proofs*, May 2017. <http://isa-afp.org/entries/CryptHOL.html>, Formal proof development.
- [13] A. Lochbihler, S. R. Sefidgar, and B. Bhatt. Game-based cryptography in hol. *Archive of Formal Proofs*, 2017. http://isa-afp.org/entries/Game_Based_Crypto.shtml, Formal proof development.
- [14] J. C. Mitchell. Representation independence and data abstraction. In *ACM SIGACT-SIGPLAN Symposium on Principles of Programming Languages (POPL 1986), Proceedings*, pages 263–276, 1986.
- [15] T. Nipkow, L. C. Paulson, and M. Wenzel. *Isabelle/HOL - A Proof Assistant for Higher-Order Logic*, volume 2283 of *Lecture Notes in Computer Science*. Springer, 2002.
- [16] A. Petcher and G. Morrisett. The foundational cryptography framework. In *POST 2015*, volume 9036 of *LNCS*, pages 53–72. Springer, 2015.

- [17] J. Schneider, M. Eberl, and A. Lochbihler. Monad normalisation. *Archive of Formal Proofs*, May 2017. http://isa-afp.org/entries/Monad_Normalisation.html, Formal proof development.
- [18] G. Shafi and S. Micali. Probabilistic encryption. *Journal of Computer and System Sciences*, 28(2):270–299, 1984.
- [19] V. Shoup. Sequences of games: A tool for taming complexity in security proofs. Cryptology ePrint Archive, Report 2004/332, 2004.
- [20] P. Wadler and S. Blott. How to make ad-hoc polymorphism less ad hoc. In *ACM SIGPLAN-SIGACT Symposium on Principles of Programming Languages (POPL 1989), Proceedings*, pages 60–76, 1989.
- [21] M. Wenzel. Isar — a generic interpretative approach to readable formal proof documents. In *International Conference on Theorem Proving in Higher Order Logics (TPHOL 1999), Proceedings*, pages 167–183, 1999.